Cálculo Diferencial e Integral: um KIT de sobrevivência



This woksheet is in Portuguese language.

Introdução ao Maple

Prof. Doherty Andrade - doherty200@hotmail.com

0. **Introdução -** O que é o Maple?

O Maple é mais um sistema de computação algébrica como o Mathematica, Derive, Mupad e MathLab (e para os mais velhos o Reduce). Computação Algébrica é sinônimo de computação simbólica. No Maple é possível efetuarmos operações simbólicas e cálculos abstratos de uma maneira simples. O Maple na verdade é uma linguagem de programação que tem se mostrado a cada dia mais eficiente. É uma tendência cada dia mais forte o uso de computação algébrica no ensino de ciências exatas, os estudantes só têm a ganhar com esta ferramenta. Aqueles que resistirem a a esta idéia podem ficar em situação de desvantagem neste mundo globalizado.

Estas notas foram organizadas para servirem de apoio para iniciantes. Para aproveitar o máximo leia, refaça e modifique os exemplos e sobretudo aprenda a usar o help.

Visite regularmente o site KIT de sobrevivência em cálculo **www.dma.uem.br/kit**

No menu clique em **Tools** e aprenda a usar os novos recursos do Maple.
Visite o site para mais aplicaçoes http://www.maplesoft.com/applications/

1.**Primeira parte -** Introdução com os principais comandos

Nesta sessão de trabalho veremos os principais
recursos do MapleV para operações algébricas
(numericamente e simbolicamente).

Não esqueça: após uma instrução teclar "ponto e vírgula" ou "dois pontos".
Descubra a diferença entre eles.

```
> # este é o símbolo de comentário. O Maple ignora tudo que
  vem a seguir.   # Símbolos de operações +,*, -,/,**=^
```

1. MANIPULANDO NÚMEROS

Faça a seguinte conta .

```
> 32*12^13;  # APERTE [enter]
```

$$3423782572130304 \tag{2.1}$$

```
> # Calcular o fatorial de 20
> 20!;
```

$$2432902008176640000 \tag{2.2}$$

```
> # Decompor o número acima em fatores primos
> ifactor(%);
```

$$(2)^{18} (3)^{8} (5)^{4} (7)^{2} (11) (13) (17) (19) \tag{2.3}$$

```
> # Expandir o número acima
> expand(%);
```

$$2432902008176640000 \tag{2.4}$$

Trabalhando com aritmética EXATA.

```
> (2^30/3^20)*sqrt(3);
```

$$\frac{1073741824}{3486784401} \sqrt{3} \tag{2.5}$$

Não apareceu nenhum arredondamento. O valor é exato. Para ver uma aproximação em decimais executamos o evalf (**evaluation with floating point**)

```
> evalf(%);
```

$$0.5333783739 \tag{2.6}$$

Outro exemplo:

```
> sin(sqrt(2/3)*Pi);
```

$$\sin\left(\frac{1}{3} \sqrt{6}\, \pi\right) \tag{2.7}$$

```
> evalf(%);
```

$$0.5450870920 \tag{2.8}$$

Outro exemplo:

```
> 20*root[5](Pi);
```

$$20\, \pi^{1/5} \tag{2.9}$$

```
> # Quantos dígitos usados?
> Digits;
```

$$10 \tag{2.10}$$

Se você quer a resposta com mais dígitos, é só pedir.

```
> Digits:= 30: evalf(sin(sqrt(2/3)*Pi));
```

$$0.545087092563010770211034449071 \tag{2.11}$$

Voltando a 10 digitos.

```
> Digits:= 10:
```

Fazendo somatórios. Descubra a diferença entre os comandos **Sum** e **sum**

```
> Sum((1+i)/(1+i^4), i=1..10);
```

$$\sum_{i=1}^{10} \frac{1+i}{i^4+1} \tag{2.12}$$

```
> value(%);
```

$$\frac{51508056727594732913722}{40626648938819200088497} \tag{2.13}$$

```
> Sum(1/n^2,n=1..40);
```

$$\sum_{n=1}^{40} \frac{1}{n^2} \tag{2.14}$$

```
> value(%);
```

$$\frac{4625296921049975441542742158630 9}{2854691655487548938516879424000 0} \tag{2.15}$$

```
> evalf(%);
```

$$1.620243963 \tag{2.16}$$

Produtórios, como no somatório, descubra a diferença entre os comandos **Product** e **product**

```
> Product(((i^2+3*i-11)/(i+3)), i=0..10);
```

$$\prod_{i=0}^{10} \left(\frac{i^2+3\,i-11}{i+3} \right) \tag{2.17}$$

```
> value(%);
```

$$-\frac{7781706512657}{40435200} \tag{2.18}$$

```
> evalf(%,50);
```

$$-1.9244881965854008388730610952833175055397277619500\ 10^5 \tag{2.19}$$

Repita os comandos de somatórios e produtórios, iniciando com letra minúscula para ver o que ocorre.

```
> sum((1+i)/(1+i^4), i=1..10);
```

$$\frac{51508056727594732913722}{40626648938819200088497} \tag{2.20}$$

```
> product(((i^2+3*i-11)/(i+3)), i=0..10);
```

$$-\frac{7781706512657}{40435200} \tag{2.21}$$

Números Complexos:

```
> (3+5*I)/(7+4*I); # dividindo ...
```

$$\frac{41}{65}+\frac{23}{65}\,\mathrm{I} \tag{2.22}$$

```
> (2-4*I)*(1+I);
```

$$6-2\,\mathrm{I} \tag{2.23}$$

```
> sqrt(4*I);
```

$$\sqrt{2}+\mathrm{I}\sqrt{2} \tag{2.24}$$

```
> (sqrt(4*I))^2;
```

$$\left(\sqrt{2}+\mathrm{I}\sqrt{2}\right)^2 \tag{2.25}$$

```
> evalc(%);
```

$$4\,\mathrm{I} \tag{2.26}$$

Funções Famosas e constantes famosas:

```
> evalf(exp(1),40); #constante e com 40 algarismos
```

$$2.718281828459045235360287471352662497757 \tag{2.27}$$

```
> evalf(Pi,50);  #constante Pi com 50 algarismos
```

$$3.1415926535897932384626433832795028841971693993751 \tag{2.28}$$

O Maple é poderoso em cálculos simbólicos.
Operações Simbólicas:

```
> (x+y)^3*(x+y)^2;
```

$$(x+y)^5 \tag{2.29}$$

```
> expand(%);
```

$$x^5+5\,x^4\,y+10\,x^3\,y^2+10\,x^2\,y^3+5\,x\,y^4+y^5 \tag{2.30}$$

Fatorar o polinômio acima

```
> factor(%);
```

$$(x+y)^5 \tag{2.31}$$

```
> simplify(cos(x)^5 + sin(x)^4 + 2*cos(x)^2 - 2*sin(x)^2 - cos
  (2*x));
```

$$\cos(x)^4\,(\cos(x)+1) \tag{2.32}$$

Normalizar formas racionais

```
> normal((x^3-y^3)/(x^2+x-y-y^2));
```

$$\frac{x^2+x\,y+y^2}{x+1+y} \tag{2.33}$$

2.Trabalhando com nomes:

```
> A:= (41*x^2+x+1)^2;
```

$$A := \left(41\,x^2+x+1\right)^2 \tag{2.34}$$

```
> B:= expand(A);
```

$$B := 1681\,x^4+82\,x^3+83\,x^2+2\,x+1 \tag{2.35}$$

```
> C:= 32*x^3-4;
```

$$C := 32\,x^3-4 \tag{2.36}$$

```
> # Dividir A por C
```

```
> A/C;
```

(2.37)

$$\frac{\left(41\,x^2+x+1\right)^2}{32\,x^3-4} \tag{2.37}$$

Escrever A/C em fraçoes parciais.

```
> convert(A/C, parfrac,x);
```

$$\frac{1681}{32}\,x+\frac{41}{16}+\frac{2209}{192\,(2\,x-1)}+\frac{1}{192}\,\frac{-2426\,x+1669}{4\,x^2+2\,x+1} \tag{2.38}$$

Escrever a função cot em termos de exponenciais.

```
> convert(cot(x) , exp);
```

$$\frac{\mathrm{I}\left(\mathrm{e}^{\mathrm{I}x}+\mathrm{e}^{-\mathrm{I}x}\right)}{\mathrm{e}^{\mathrm{I}x}-\mathrm{e}^{-\mathrm{I}x}} \tag{2.39}$$

Mais duas coisas importantes, raiz de polinomio com a multiplicidade.

```
> solve(a*x^2+b*x+c,x);
```

$$\frac{1}{2}\,\frac{-b+\sqrt{-4\,a\,c+b^2}}{a},\ -\frac{1}{2}\,\frac{b+\sqrt{-4\,a\,c+b^2}}{a} \tag{2.40}$$

```
> polinomio:=9*x^3-37*x^2+47*x-19;
```

$$polinomio := 9\,x^3-37\,x^2+47\,x-19 \tag{2.41}$$

```
> solve(polinomio,x);
```

$$\frac{19}{9},\ 1,\ 1 \tag{2.42}$$

Derivando um polinômio

```
> p:=x->9*x^3-37*x^2+47*x-19;
```

$$p := x \rightarrow 9\,x^3+VectorCalculus{:-}\text{`-`}\left(37\,x^2\right)+47\,x+(-19) \tag{2.43}$$

```
> D(p);
```

$$x \rightarrow 27\,x^2-74\,x+47 \tag{2.44}$$

```
> D(p)(1);
```

$$0 \tag{2.45}$$

Para determinar as raizes (reais ou complexas) numericamente de um polinômio

```
> fsolve(p(x)=0,x);
```

$$1.,\ 1.,\ 2.111111111 \tag{2.46}$$

Resolvendo um sistema

```
> eqns := {7*x-14*y=1.111, -14*x+140*y=-0.642};
```

$$eqns := \{-14\,x+140\,y=-0.642,\ 7\,x-14\,y=1.111\} \tag{2.47}$$

```
> sols := solve( eqns );
```

$$sols := \{x=0.1869285714,\ y=0.01410714286\} \tag{2.48}$$

Como usar o HELP
Você pode ver quais são os pacotes do Maple e seus comandos **consultando o help.**
O help do Maple é muito bom. É importante saber usá-lo. O help tem três níveis de ajuda com um `?questao` ou com dois `??questao` ou com três `???questao.` Descubra a diferença entre eles.

Para ver os pacotes do Maple digite `?index[package];`

```
> ?index[package];
```

Cada pacote é carregado com o comando **with**.

```
> #with(student);       #exemplo
> #?index[functions];   #exemplo
> #with(linalg);        #exemplo
```

2. Segunda parte - Noções de cálculo diferencial e integral

Noções de Cálculo diferencial e integral com uma variável

```
> # Simbolo para comentário
```

1 - CALCULANDO LIMITES

Definindo uma função de duas variáveis.

```
> f1:=(x,y) -> (x^2-5*y)/(x^3+2*x);
```

$$f1 := (x, y) \rightarrow \left(x^2 + VectorCalculus{:-}\grave{}\text{-}\grave{}(5\,y)\right)\frac{1}{x^3 + 2\,x} \qquad (3.1)$$

Podemos definir também uma função cuja expressão muda com as condiçoes no argumento. Por exemplo:

```
> f:=x ->piecewise(1<=x and x<2, x^2,2*x+10);
```

$$f := x \rightarrow piecewise\left(1 \leq x \textbf{ and } x < 2, x^2, 2\,x + 10\right) \qquad (3.2)$$

```
> f(x);
```

$$\begin{cases} x^2 & 1 \leq x \textbf{ and } x < 2 \\ 2\,x + 10 & otherwise \end{cases} \qquad (3.3)$$

```
> plot(f(x),x=-3..3);
```



Calendo limites.

```
> limit(f1(x,y),x=1);
```

$$-\frac{5}{3}y+\frac{1}{3} \tag{3.4}$$

```
> Limit(x*cos(1/x^2), x=0, left); #um limite dificil??
```

$$\lim_{x\to 0^-} x\cos\left(\frac{1}{x^2}\right) \tag{3.5}$$

```
> value(%);
```

$$0 \tag{3.6}$$

Em limites, nem tudo são flores.
Vejamos alguns exemplos em que o Maple não conseque dar uma resposta satisfatória.
E exemplos onde o limite é calculado em vários pontos.

```
> F2:=(x,y,z)-> (x^2+y^2+z^2)/ (x^2+y^2+z^2+1);
```

$$F2 := (x,y,z) \rightarrow (x^2+y^2+z^2)\,\frac{1}{x^2+y^2+z^2+1} \tag{3.7}$$

```
> limit(F2(x,y,z),{x=1,y=2,z=3});### observe, o ponto está
  entre chaves.
```

$$\frac{14}{15} \tag{3.8}$$

```
> limit((x^2-y^2)/(x^2+y^2), {x=0,y=0});
```

$$\textit{undefined} \tag{3.9}$$

```
> limit(sin(x^2+y^2)/(x^2+y^2),{x=0,y=0});
```

$$\mathit{limit}\left(\frac{\sin\left(x^2+y^2\right)}{x^2+y^2},\ \{x=0, y=0\}\right) \tag{3.10}$$

Vamos ver o **gráfico** de uma função. Tomemos como exemplo a função $f(x) = \dfrac{x^2-5}{x^2+1}$, no intervalo [-2,2].

```
> plot((x^2-5)/(x^2+1),x=-2..2);
```



Faça outros gráficos. Experimente funções conhecidas por você tal como seno, cosseno,etc...

Outra forma de definir uma função é por meio do chamado **procedimento**.

Um procedimento tem a forma:

```
P : = proc(argumentos)
     local (variáveis locais)
    instruções a serem executadas
    end;
```

Vamos ver um exemplo, mais tarde voltaremos neste ponto com mais cuidado.
Neste exemplo, definimos a função salto

```
> salto:=proc(x);
if 1<=x then x^2+1 else cos(x) fi;
end;
```

$$salto := \mathbf{proc}(x)\ \mathbf{if}\ 1 <= x\ \mathbf{then}\ x\hat{}2+1\ \mathbf{else}\ \cos(x)\ \mathbf{end\ if}\ \mathbf{end\ proc} \quad (3.11)$$

```
> salto(-10);
```

$$\cos(10) \quad (3.12)$$

```
> salto(2.2);
```

$$5.84 \quad (3.13)$$

Outro exemplo.

```
> pp:=proc(x); x^3;end;
```

$$pp := \mathbf{proc}(x)\ x\hat{}3\ \mathbf{end\ proc} \quad (3.14)$$

```
> pp(sqrt(2));
```

$$2\sqrt{2} \quad (3.15)$$

```
> pp((2)^(1/3));
```

$$2 \quad (3.16)$$

Agora vamos voltar aos limites e escrever Limit no lugar de limit .

```
> Limit(f1(x,2), x=1);
```

$$\lim_{x \to 1} \frac{x^2-10}{x^3+2x} \quad (3.17)$$

```
> value(%);
```

$$-3 \quad (3.18)$$

Se um comando começa com letra maiúscula, então ele é "inerte".
Isto é, só escreve mas não calcula.

```
> A:=sin(2*x)/x;
```

$$A := \frac{\sin(2x)}{x} \quad (3.19)$$

Observar que A não é do tipo A:=x -> sin(2*x)/x, função
Portanto, A não é função (é uma expressão).

```
> Limit(A, x=0);
```

$$\lim_{x \to 0} \frac{\sin(2x)}{x} \quad (3.20)$$

```
> value(%);
```

$$2 \quad (3.21)$$

```
> Limit(A, x=infinity);
```

$$\lim_{x \to \infty} \frac{\sin(2x)}{x} \quad (3.22)$$

```
> value(%);
```

$$0 \quad (3.23)$$

Lá vão dois limites infinitos.

```
> Limit(1/x, x=0, left);
```

$$\lim_{x \to 0^-} \frac{1}{x} \tag{3.24}$$

```
> value(%);
```

$$-\infty \tag{3.25}$$

Abaixo, usamos "limit" com l minúscula.

```
> limit(1/x, x=0, right);
```

$$\infty \tag{3.26}$$

```
> limit(1/x, x=0, left);
```

$$-\infty \tag{3.27}$$

O Maple pode calcular limites para funções de mais de uma variável . Vamos ver exemplos.

```
> limit(x^2-y^3+z^4,{x=1,y=2,z=3});
```

$$74 \tag{3.28}$$

```
> limit(x^2/(x^2+y^2+z^2+1),{x=1,y=2});
```

$$\frac{1}{z^2+6} \tag{3.29}$$

```
> limit(limit(x/(x^2+y^2),{x=y^2}),y=0);
```

$$1 \tag{3.30}$$

2 - **CALCULANDO ALGUMAS DERIVADAS**

```
> f2:=x -> sin(2*x);
```

$$f2 := x \to \sin(2\,x) \tag{3.31}$$

Derivando uma vez,

```
> diff(f2(x),x);
```

$$2\cos(2\,x) \tag{3.32}$$

Derivando 3 vezes,

```
> diff(f2(x),x,x,x);
```

$$-8\cos(2\,x) \tag{3.33}$$

Observe que para derivar 3 vezes podemos usar x, x, ou $3.

```
> diff(f2(x),x$3);
```

$$-8\cos(2\,x) \tag{3.34}$$

Calculando derivadas com operador diferencial "D". A saída (output) é sempre uma função

```
> h:=x -> x^2;
```

$$h := x \to x^2 \tag{3.35}$$

```
> h(x);
```

$$x^2 \tag{3.36}$$

```
> D(h);  # Vai sair uma função !
```

$$x \to 2\,x \tag{3.37}$$

```
> D(h)(x);# aqui calculamos a derivada no ponto x
```

$$2\,x \tag{3.38}$$

```
> D(h)(1);# aqui calculamos a derivada no ponto x=1
```

$$2 \tag{3.39}$$

```
> diff(R(x)*S(x),x);  # derivada do produto
```

(3.40)

$$\left(\frac{d}{dx} R(x)\right) S(x) + R(x) \left(\frac{d}{dx} S(x)\right) \quad \mathbf{(3.40)}$$

```
> diff(T(x,y(x)),x); # derivada da composta de T(x,y(x))
```

$$D_1(T)(x, y(x)) + D_2(T)(x, y(x)) \left(\frac{d}{dx} y(x)\right) \quad \mathbf{(3.41)}$$

3 - UM PROBLEMINHA DE CÁLCULO INTERESSANTE.

Dado um ponto P = (a,b) e o gráfico de uma função f, calcular a distância de P ao gráfico da f.

Ao começar um novo problema costuma-se "zerar" a memória. Fazemos isto com o comando abaixo.

```
> restart:
```

Por definição, a distância de P ao gráfico da f é a menor distância entre os pontos Q do gráfico e o ponto P. Vamos então definir:

distância=|| P-Q ||

Neste exemplo tomamos e P=(2,3). Um ponto do gráfico de f é da forma (x, cos(x)), assim a distância do ponto P a um ponto do gráfico é dada por

```
> dist:=sqrt( (2-x)^2 + (3-cos(x))^2 );
```

$$dist := \sqrt{(2-x)^2 + (3-\cos(x))^2} \quad \mathbf{(3.42)}$$

Queremos a menor distância. Da fórmula acima vemos que dist é uma função só de x. Portanto, do Cálculo, concluímos que a menor distância deve ser dada por um x que anula a derivada de dist.

Vamos ver o gráfico da função dist.

```
> plot( sqrt( (2-x)^2 + (3-cos(x))^2),x=0..2);
```




```
> derivada:=diff(dist,x);
```

$$derivada := \frac{1}{2}\,\frac{2\,x-4+2\,(3-\cos(x))\,\sin(x)}{\sqrt{(2-x)^2+(3-\cos(x))^2}} \tag{3.43}$$

Resolvendo a equação derivada=0 para "x" em [0,2]. Já vi o gráfico.

```
> x[min]:=fsolve(derivada, x ,0..2 );
```

$$x_{\min} := 0.6551969516 \tag{3.44}$$

```
> evalf(%);
```

$$0.6551969516 \tag{3.45}$$

Escrevendo a resposta utilizando o comando subs. (Substituir o valor de x por na expressão dist).

```
> menor[dist]:=subs(x=x[min], dist);
```

$$menor_{\sqrt{(2-x)^2+(3-\cos(x))^2}} := \sqrt{1.808495238+(3-\cos(0.6551969516))^2} \tag{3.46}$$

```
> resposta:=evalf(%);
```

$$resposta := 2.584504271 \tag{3.47}$$

```
> # Descubra o que foi feito (abaixo)
```

```
> L:=[ [2,3], [x[min] , cos(x[min]) ]];
```

$$L := [[2, 3], [0.6551969516, 0.7929279378]] \tag{3.48}$$

```
> plot({cos(x), L }, x=0..3 , scaling=constrained);
```



4 - **SÉRIES DE TAYLOR**

A função "series" escreve a 'série de Taylor de funções analíticas. Em geral, a resposta é dada em termos de uma expansão de ordem 6.

```
> S:=exp(x)*x^2;
```

$$S := e^x x^2 \tag{3.49}$$

```
> S1:=series( S, x=0 ); # Em torno de x=0.
```

$$S1 := x^2 + x^3 + \frac{1}{2}\, x^4 + \frac{1}{6}\, x^5 + O(x^6) \tag{3.50}$$

Queremos agora uma expansão de ordem 10.

```
> S2:=series( S , x=0, 10 );
```

$$S2 := x^2 + x^3 + \frac{1}{2}\, x^4 + \frac{1}{6}\, x^5 + \frac{1}{24}\, x^6 + \frac{1}{120}\, x^7 + \frac{1}{720}\, x^8 + \frac{1}{5040}\, x^9 + O(x^{10}) \tag{3.51}$$

Converter a série num polinômio.

```
> S3:=convert(S2,  polynom );
```

$$S3 := x^2 + x^3 + \frac{1}{2}\,x^4 + \frac{1}{6}\,x^5 + \frac{1}{24}\,x^6 + \frac{1}{120}\,x^7 + \frac{1}{720}\,x^8 + \frac{1}{5040}\,x^9 \tag{3.52}$$

```
> S3(2);      # inútil
```

$$\mathit{table}([\min = 0.6551969516])(2)^2 + \mathit{table}([\min = 0.6551969516])(2)^3 + \frac{1}{2}\,\mathit{table}([\min = 0.6551969516])(2)^4 + \frac{1}{6}\,\mathit{table}([\min = 0.6551969516])(2)^5 + \frac{1}{24}\,\mathit{table}([\min = 0.6551969516])(2)^6 + \frac{1}{120}\,\mathit{table}([\min = 0.6551969516])(2)^7 + \frac{1}{720}\,\mathit{table}([\min = 0.6551969516])(2)^8 + \frac{1}{5040}\,\mathit{table}([\min = 0.6551969516])(2)^9 \tag{3.53}$$

```
> value(%);  #inútil
```

$$\mathit{table}([\min = 0.6551969516])(2)^2 + \mathit{table}([\min = 0.6551969516])(2)^3 + \frac{1}{2}\,\mathit{table}([\min = 0.6551969516])(2)^4 + \frac{1}{6}\,\mathit{table}([\min = 0.6551969516])(2)^5 + \frac{1}{24}\,\mathit{table}([\min = 0.6551969516])(2)^6 + \frac{1}{120}\,\mathit{table}([\min = 0.6551969516])(2)^7 + \frac{1}{720}\,\mathit{table}([\min = 0.6551969516])(2)^8 + \frac{1}{5040}\,\mathit{table}([\min = 0.6551969516])(2)^9 \tag{3.54}$$

Vamos transformar a expressão S3 numa função de verdade.

```
> P:=unapply( S3 ,  x );
```

$$P := x \rightarrow x^2 + x^3 + \frac{1}{2}\,x^4 + \frac{1}{6}\,x^5 + \frac{1}{24}\,x^6 + \frac{1}{120}\,x^7 + \frac{1}{720}\,x^8 + \frac{1}{5040}\,x^9 \tag{3.55}$$

```
> evalf( P(1) );
```

$$2.718253968 \tag{3.56}$$

```
> evalf( subs(x=1, S) );
```

$$2.718281828 \tag{3.57}$$

5 - CALCULANDO INTEGRAIS

```
> Int(ln(x),x);        # Com I maiúscula
```

$$\int \ln(x)\,dx \tag{3.58}$$

```
> int(ln(x),x);        # Com i minúscula
```

$$x\ln(x) - x \tag{3.59}$$

```
> int(tan(x),x);        # Com I minúscula
```

$$-\ln(\cos(x)) \tag{3.60}$$

```
> Int(1/ sqrt(1-x^2),x);
```

$$\int \frac{1}{\sqrt{-x^2+1}}\,dx \tag{3.61}$$

```
> value(%);
```

$$\arcsin(x) \tag{3.62}$$

```
> diff(%,x);
```

$$\frac{1}{\sqrt{-x^2+1}} \tag{3.63}$$

```
> AA:=Int(x^2*exp(x^2),x=0..1);
```

$$AA := \int_0^1 x^2 e^{x^2} \, dx \tag{3.64}$$

```
> resp:=value(AA);
```

$$resp := \frac{1}{2} e - \frac{1}{4} \sqrt{\pi} \operatorname{erfi}(1) \tag{3.65}$$

Maple utilizou a função erro "erf" que é definida por

ou o que o mesmo: $\operatorname{erf}(x) = \frac{2\left(\int_0^x e^{-t^2} \, dt\right)}{\sqrt{\pi}}$

Mas mesmo assim podemos ter uma aproximação da integral utilizando-se evalf(AA).

```
> evalf(AA);
```

$$0.6278150413 \tag{3.66}$$

Os próximos exemplos são conhecidos.

Vamos ver alguma coisa sobre Integrais múltiplas.

```
> Int( Int(x^2+y^2, y=0..2 ),x=0..2 );
```

$$\int_0^2 \int_0^2 (x^2+y^2) \, dy \, dx \tag{3.67}$$

```
> value(%);
```

$$\frac{32}{3} \tag{3.68}$$

```
> Int( Int( Int( z*cos(x), x=0..2*y ), y=0..2 ), z=1..2 );
```

$$\int_1^2 \int_0^2 \int_0^{2y} z \cos(x) \, dx \, dy \, dz \tag{3.69}$$

```
> value(%);
```

$$\frac{3}{4} - \frac{3}{4} \cos(4) \tag{3.70}$$

```
> evalf(%);
```

$$1.240232716 \tag{3.71}$$

Veremos alguns exemplos de integrais improprias.

```
> f8:= exp(-u*x)*ln(x)*sqrt(x);
```

(3.72)

$$f8 := e^{-u\,x} \ln(x) \sqrt{x} \qquad \textbf{(3.72)}$$

```
> Int(f8,x=0..infinity);
```

$$\int_0^{\infty} e^{-u\,x} \ln(x) \sqrt{x}\, dx \qquad \textbf{(3.73)}$$

Como nada é conhecido sobre u, MapleV não pode determinar uma resposta. Podemos usar o comando ASSUME para informar ao MapleV sobre u.

```
> assume(u<0); int(f8,x=0..infinity);
```

$$\infty \qquad \textbf{(3.74)}$$

Esta integral diverge. Por outro lado, se $u > 0$ temos convergência e uma resposta simbólica.

```
> assume(u>0);int(f8,x=0..infinity);
```

$$\frac{1}{2} \frac{\sqrt{\pi}\,(2 - \gamma - \ln(4\,u\sim))}{u\sim^{3/2}} \qquad \textbf{(3.75)}$$

```
> Int(1/x^2 , x=1..infinity);
```

$$\int_1^{\infty} \frac{1}{x^2}\, dx \qquad \textbf{(3.76)}$$

```
> value(%);
```

$$1 \qquad \textbf{(3.77)}$$

Outro exemplo.

```
> int(1/x, x=0..1);
```

$$\infty \qquad \textbf{(3.78)}$$

Mais um exemplo simples.

```
> a:=2; b:=4; Int(1/x^3,x=a..b)=int(1/(x^3), x = a..b);
```

$$a := 2$$
$$b := 4$$
$$\int_2^{4} \frac{1}{x^3}\, dx = \frac{3}{32} \qquad \textbf{(3.79)}$$

Valor principal de Cauchy: peça ao Maple.

```
> int(1/(x^3), x=-1..4, 'CauchyPrincipalValue' );
```

$$\frac{15}{32} \qquad \textbf{(3.80)}$$

Um exemplo sobre transformada de Laplace.
Vamos usar o pacote **inttrans**.

```
> restart; with(inttrans):
> laplace(t**3-cos(t)=y(t), t, s);
```

$$\frac{6}{s^4} - \frac{s}{s^2+1} = laplace(y(t), t, s) \qquad \textbf{(3.81)}$$

```
> laplace(t^(1/2)-exp(-t)+sinh(a*t), t, s);
```

(3.82)

$$\frac{1}{2}\frac{\sqrt{\pi}}{s^{3/2}} - \frac{1}{s+1} + \frac{a}{-a^2+s^2} \tag{3.82}$$

3.**Terceira parte** - **Plotando gráficos**

Vamos iniciar com gráficos simples.
As vezes não é necessário escrever o argumento x

```
> plot(sin,-Pi..Pi);
```



Plotar o gráfico de com x em [0,2Pi]
É necessario escrever o argumento x

```
> plot(sin(2*x), x=0..2*Pi);
```



Vamos plotar um gráfico com descontinuidade:

```
> plot(1/(x-1), x=-3..3, -5..5, discont=true);
```



Acrescentando o titulo ao seu grafico. Usamos a opção "title"

```
> plot(ln(x), x=0..3 , title=`Logaritmo Natural`);
```



Usando opção "scaling=constrainded": escala 1-1.

```
> plot(exp(-x^2), x=-2..2 , scaling=constrained);
```



Podemos plotar dois ou mais gráficos juntos. Este é um recurso muito util.
Juntando dois Gráficos: use chaves- plot({f(x),g(x)}, x=a..b)

```
> plot({arctan, sqrt}, 0..5);
```



Plotando uma curva parametrizada: use colchetes
com t em [0,2Pi]

```
> plot( [cos(t), sin(2*t), t=0..2*Pi] );
```



Um recurso do Maple é a **animação** de gráficos.
Vamos animar um gráfico ? Parece um video- cassete.
Depois de feito o gráfico, clique em cima do desenho para iniciar a animação.

```
> with(plots):
> animate( x^2*sin(x*t),x=-10..10,t=1..2,frames=50);
```



Outra opção de plot animado.

```
> plots[animate]( plot, [sin(t), t=0..x], x=0..4*Pi );
```




```
> animate(plot,[A*x^2,x=-4..4],A=-3..3,trace=5,frames=50);
```



Os procedimentos em "plot" requerem bastante recursos de máquina. É costume "zerar a memória" após 10 plots.

```
> restart;
```

Dois Problemas interessantes.

PROBLEMA 1

A cardióide é uma curva parametrizada em coordenadas polares. Consulte o help do plot para plotar a cardióide .

```
> plot([5*(1-cos(t)) ,t , t=0..2*Pi],  coords=polar);
```



PROBLEMA 2

Um certo algorítmo gerou uma lista de pontos x
e uma lista de pontos y abaixo

```
> X:=[seq( .1*k , k=0..10 ) ]:
> Y:=[seq( 5*( (.1*k)^2-(.1*k)^3 ) , k=0..10 ) ]:
> L:=[ seq( [ X[k], Y[k] ], k=1..11) ];
```

$L := [[0., 0.], [0.1, 0.045], [0.2, 0.160], [0.3, 0.315], [0.4, 0.480], [0.5, 0.625], [0.6, 0.720], [0.7, 0.735], [0.8, 0.640], [0.9, 0.405], [1.0, 0.]]$ **(4.1)**

```
> plot(L, scaling=constrained, title=`plotando dados`);
```

plotando dados



Veremos agora gráficos mais complicados.
Plotando gráficos tridimensionais

```
> restart;
```

Para plotar gráficos em três dimensões precisamos chamar o pacote para gráficos, fazemos isto digitando "with(plots)".O comando é plot3d. Clique sobre o desenho e gire-o. Explore os recursos do Maple.

```
> with(plots):
```

```
> plot3d(sin(x^2+y^2)*cos(x^2+y^2),x=-1..1,y=-1..1);
```




```
> plot3d(-3.6+0.3599*x-0.0027*x^2-0.00166*y,x=15..107,y=0..70,
  title=`akira`);
```




```
> plot3d(exp(-x^2-y^2),x=-2..2,y=-2..2);
```




```
> animate3d({cos(t*x)*sin(t*y),-cos(t*x)*sin(t*y)},x=-Pi..Pi,
  y=-Pi..Pi,t=1..2);
```




```
> animate3d(x*cos(t*u),x=1..3,t=1..4,u=2..4,coords=spherical);
```



Mais um exemplo

```
> restart;
> with(plots):
> f:=(x,y) -> -(x^2+y^2);
```

$$f := (x, y) \rightarrow -x^2 - y^2 \tag{4.2}$$

```
> plot3d(f(x,y), x=-2..2, y=-2..2, scaling=constrained);
```



Plotando algo mais bonito

```
> plot3d(sin(x)*cos(y), x=-Pi..Pi, y=-Pi..Pi );
```



Observe as funções especiais que esse pacote tem.
Para se saber sobre alguma delas execute ?nome,ou ??nome ou ???nome.

Veremos exemplos de "gradplot",

```
> f :=(x,y) -> x*exp(-x^2-y^2);
```

$$f := (x, y) \rightarrow x\, e^{-x^2 - y^2} \tag{4.3}$$

```
> plot3d( f(x,y), x=-2..2, y=-2..2);
```



Vamos ver o CAMPO GRADIENTE da f com "gradplot" e "countourplot"

```
> contourplot(sin(x^2+y^2),x=-1..1,y=-1..1);
```




```
> gradplot(f(x,y), x=-2..2, y=-2..2);
```




```
> gradplot3d(x^2+2*y^2+z+1,x=-1..1,y=-1..1,z=-1..1);
```



Vamos desenhar a esfera, dada implicitamente.

```
> implicitplot3d(x^2+y^2+z^2=4, x=-2..2, y=-2..2, z=-2..2 ,
  scaling=constrained );
```



Coordenadas cilindricas

```
> cylinderplot(z+ 3*cos(2*theta),theta=0..Pi,z=0..3);
```




```
> f := (5*cos(y)^2 -1)/3;
```

$$f := \frac{5}{3}\cos(y)^2 - \frac{1}{3} \tag{4.4}$$

```
> cylinderplot(f, x=0..2*Pi,y=-Pi..Pi,style=PATCH, color = f);
```



Um exemplo tubular com "tubeplot"

```
> c:=[(t- 5*Pi)*sin(t)/3,(t-5*Pi)*cos(t)/3,(t-5*Pi)*.9, t=0.
  .5*Pi]:
> tubeplot( c, radius=(t-5*Pi)*.2, orientation=[-37,81],
  tubepoints=25, style=hidden);
```



Exercicios

```
> # Leia "?plots[spacecurve]" e desenhe uma espiral.
> # DICA: use a opção "color=black"
> # plot([5*(1-cos(t)) ,t , t=0..2*Pi],  coords=polar);
> # X:=[seq( .1*k , k=0..10 ) ]:
> # Y:=[seq( 5*( (.1*k)^2-(.1*k)^3 ) , k=0..10 ) ]:
> # L:=[ seq( [ X[k], Y[k] ], k=1..11) ];
> # plot(L, scaling=constrained,title=`plotando dados`);
```

4. **Quarta parte** - Álgebra linear

Para acessar as funções relativas a Álgebra Linear, devemos carregar a Biblioteca de Funções "pacote" linalg.

```
> restart:
> with(linalg):
```

Para se saber mais sobre qualquer uma da funções acima listada, pode-se consultar o help iterativo. Por exemplo, para se saber mais sobre "charmat": executa-se ?charmat;

Você pode por um, dois ou três sinais de ?. O que cada um deles te dá?

1 - DEFINIDO VETORES E MATRIZES
(Usando "vector", "matrix" e "array")

```
> u:=vector( [2,sin(x),4] ); ### ou
```

$$u := \begin{bmatrix} 2 & \sin(x) & 4 \end{bmatrix} \quad (5.1)$$

```
> uu:=vector(3,[x,y,z]);
```

$$uu := \begin{bmatrix} x & y & z \end{bmatrix} \quad (5.2)$$

```
> v:=array( [ [1,1-sin(x),x] ] );
```

$$v := \begin{bmatrix} 1 & 1-\sin(x) & x \end{bmatrix} \quad (5.3)$$

```
> vv:=convert(v, vector);
```

$$vv := \begin{bmatrix} 1 & 1-\sin(x) & x \end{bmatrix} \quad (5.4)$$

```
> ut:=transpose(u);   # tranposta de um vetor???
```

$$ut := linalg\text{:-}transpose(u) \quad (5.5)$$

```
> wt:=transpose(v);   # transposta de um array
```

$$wt := \begin{bmatrix} 1 \\ 1-\sin(x) \\ x \end{bmatrix} \quad (5.6)$$

```
> M:=matrix( [ [1, 2, -3], [x-3, 4, 0] , [2, 0 ,-1] ]);
```

$$M := \begin{bmatrix} 1 & 2 & -3 \\ x-3 & 4 & 0 \\ 2 & 0 & -1 \end{bmatrix} \quad (5.7)$$

```
> N:=array( [ [1, 2, -3], [x-3, 4, 0] , [2, 0 ,-1] ] );
```

$$N := \begin{bmatrix} 1 & 2 & -3 \\ x-3 & 4 & 0 \\ 2 & 0 & -1 \end{bmatrix} \quad (5.8)$$

```
> # para adicionar vetores e matrizes
> # é: "+"  e "evalm"
> u+vv; ####  o que dá?
```

$$u + vv \quad (5.9)$$

```
> evalm(%); ### avalia o resultado anterior
```

$$\begin{bmatrix} 3 & 1 & 4+x \end{bmatrix} \quad (5.10)$$

```
> MM:=matrix([[1,2,3],[0,1,-1],[0,0,1]]);
```

$$MM := \begin{bmatrix} 1 & 2 & 3 \\ 0 & 1 & -1 \\ 0 & 0 & 1 \end{bmatrix} \quad (5.11)$$

```
> NN:=evalm(2*MM);
```

$$NN := \begin{bmatrix} 2 & 4 & 6 \\ 0 & 2 & -2 \\ 0 & 0 & 2 \end{bmatrix} \tag{5.12}$$

```
> MM+NN;## o que dá?
```

$$MM + NN \tag{5.13}$$

```
> evalm(%);
```

$$\begin{bmatrix} 3 & 6 & 9 \\ 0 & 3 & -3 \\ 0 & 0 & 3 \end{bmatrix} \tag{5.14}$$

Multiplicando matrizes com o "multiply"   ou "&* "

```
> C:=multiply(MM,NN);
```

$$C := \begin{bmatrix} 2 & 8 & 8 \\ 0 & 2 & -4 \\ 0 & 0 & 2 \end{bmatrix} \tag{5.15}$$

```
> F:=evalm(MM &* NN);
```

$$F := \begin{bmatrix} 2 & 8 & 8 \\ 0 & 2 & -4 \\ 0 & 0 & 2 \end{bmatrix} \tag{5.16}$$

```
> evalm(MM^3);
```

$$\begin{bmatrix} 1 & 6 & 3 \\ 0 & 1 & -3 \\ 0 & 0 & 1 \end{bmatrix} \tag{5.17}$$

```
> evalm(MM+MM^2+MM^3);
```

$$\begin{bmatrix} 3 & 12 & 10 \\ 0 & 3 & -6 \\ 0 & 0 & 3 \end{bmatrix} \tag{5.18}$$

Mais coisas básicas...

```
> M1:=array(1..3,1..3,[[a,b,c],[1,2,3],[alpha,beta,gamma]]);
```

$$M1 := \begin{bmatrix} a & b & c \\ 1 & 2 & 3 \\ \alpha & \beta & \gamma \end{bmatrix} \tag{5.19}$$

```
> det(M1);
```

$$-3\,a\,\beta + 2\,a\,\gamma + 3\,\alpha\,b - 2\,\alpha\,c - b\,\gamma + \beta\,c \tag{5.20}$$

```
> M1_inversa:=inverse(M1);
```

$$M1\_inversa := \left[\left[\frac{-3\beta+2\gamma}{-3a\beta+2a\gamma+3\alpha b-2\alpha c-b\gamma+\beta c}, -\frac{b\gamma-\beta c}{-3a\beta+2a\gamma+3\alpha b-2\alpha c-b\gamma+\beta c}, \frac{3b-2c}{-3a\beta+2a\gamma+3\alpha b-2\alpha c-b\gamma+\beta c}\right], \left[-\frac{-3\alpha+\gamma}{-3a\beta+2a\gamma+3\alpha b-2\alpha c-b\gamma+\beta c}, \frac{a\gamma-\alpha c}{-3a\beta+2a\gamma+3\alpha b-2\alpha c-b\gamma+\beta c}, -\frac{3a-c}{-3a\beta+2a\gamma+3\alpha b-2\alpha c-b\gamma+\beta c}\right], \left[-\frac{2\alpha-\beta}{-3a\beta+2a\gamma+3\alpha b-2\alpha c-b\gamma+\beta c}, -\frac{a\beta-\alpha b}{-3a\beta+2a\gamma+3\alpha b-2\alpha c-b\gamma+\beta c}, \frac{2a-b}{-3a\beta+2a\gamma+3\alpha b-2\alpha c-b\gamma+\beta c}\right]\right] \quad (5.21)$$

Outro exemplo.

```
> M_11:=array(1..3,1..3,[[1,-1,0],[1,2,3],[2,1,0]]);
```

$$M\_11 := \begin{bmatrix} 1 & -1 & 0 \\ 1 & 2 & 3 \\ 2 & 1 & 0 \end{bmatrix} \quad (5.22)$$

```
> det(M_11);
```

$$-9 \quad (5.23)$$

```
> M_2:=inverse(M_11);
```

$$M\_2 := \begin{bmatrix} \frac{1}{3} & 0 & \frac{1}{3} \\ -\frac{2}{3} & 0 & \frac{1}{3} \\ \frac{1}{3} & \frac{1}{3} & -\frac{1}{3} \end{bmatrix} \quad (5.24)$$

```
> multiply(M1, M1_inversa);
```

$$\left[\left[\frac{a(-3\beta+2\gamma)}{-3a\beta+2a\gamma+3\alpha b-2\alpha c-b\gamma+\beta c} - \frac{b(-3\alpha+\gamma)}{-3a\beta+2a\gamma+3\alpha b-2\alpha c-b\gamma+\beta c} - \frac{c(2\alpha-\beta)}{-3a\beta+2a\gamma+3\alpha b-2\alpha c-b\gamma+\beta c}, -\frac{a(b\gamma-\beta c)}{-3a\beta+2a\gamma+3\alpha b-2\alpha c-b\gamma+\beta c} \right. \quad (5.25)$$

$$+\frac{b\,(a\,\gamma-\alpha\,c)}{-3\,a\,\beta+2\,a\,\gamma+3\,\alpha\,b-2\,\alpha\,c-b\,\gamma+\beta\,c}-\frac{c\,(a\,\beta-\alpha\,b)}{-3\,a\,\beta+2\,a\,\gamma+3\,\alpha\,b-2\,\alpha\,c-b\,\gamma+\beta\,c},$$

$$\frac{a\,(3\,b-2\,c)}{-3\,a\,\beta+2\,a\,\gamma+3\,\alpha\,b-2\,\alpha\,c-b\,\gamma+\beta\,c}-\frac{b\,(3\,a-c)}{-3\,a\,\beta+2\,a\,\gamma+3\,\alpha\,b-2\,\alpha\,c-b\,\gamma+\beta\,c}+\frac{c\,(2\,a-b)}{-3\,a\,\beta+2\,a\,\gamma+3\,\alpha\,b-2\,\alpha\,c-b\,\gamma+\beta\,c}\Bigg],$$

$$\Bigg[\frac{-3\,\beta+2\,\gamma}{-3\,a\,\beta+2\,a\,\gamma+3\,\alpha\,b-2\,\alpha\,c-b\,\gamma+\beta\,c}-\frac{2\,(-3\,\alpha+\gamma)}{-3\,a\,\beta+2\,a\,\gamma+3\,\alpha\,b-2\,\alpha\,c-b\,\gamma+\beta\,c}-\frac{3\,(2\,\alpha-\beta)}{-3\,a\,\beta+2\,a\,\gamma+3\,\alpha\,b-2\,\alpha\,c-b\,\gamma+\beta\,c},$$

$$-\frac{b\,\gamma-\beta\,c}{-3\,a\,\beta+2\,a\,\gamma+3\,\alpha\,b-2\,\alpha\,c-b\,\gamma+\beta\,c}+\frac{2\,(a\,\gamma-\alpha\,c)}{-3\,a\,\beta+2\,a\,\gamma+3\,\alpha\,b-2\,\alpha\,c-b\,\gamma+\beta\,c}-\frac{3\,(a\,\beta-\alpha\,b)}{-3\,a\,\beta+2\,a\,\gamma+3\,\alpha\,b-2\,\alpha\,c-b\,\gamma+\beta\,c},$$

$$\frac{3\,b-2\,c}{-3\,a\,\beta+2\,a\,\gamma+3\,\alpha\,b-2\,\alpha\,c-b\,\gamma+\beta\,c}-\frac{2\,(3\,a-c)}{-3\,a\,\beta+2\,a\,\gamma+3\,\alpha\,b-2\,\alpha\,c-b\,\gamma+\beta\,c}+\frac{3\,(2\,a-b)}{-3\,a\,\beta+2\,a\,\gamma+3\,\alpha\,b-2\,\alpha\,c-b\,\gamma+\beta\,c}\Bigg],$$

$$\Bigg[\frac{\alpha\,(-3\,\beta+2\,\gamma)}{-3\,a\,\beta+2\,a\,\gamma+3\,\alpha\,b-2\,\alpha\,c-b\,\gamma+\beta\,c}-\frac{\beta\,(-3\,\alpha+\gamma)}{-3\,a\,\beta+2\,a\,\gamma+3\,\alpha\,b-2\,\alpha\,c-b\,\gamma+\beta\,c}-\frac{\gamma\,(2\,\alpha-\beta)}{-3\,a\,\beta+2\,a\,\gamma+3\,\alpha\,b-2\,\alpha\,c-b\,\gamma+\beta\,c},$$

$$-\frac{\alpha\,(b\,\gamma-\beta\,c)}{-3\,a\,\beta+2\,a\,\gamma+3\,\alpha\,b-2\,\alpha\,c-b\,\gamma+\beta\,c}+\frac{\beta\,(a\,\gamma-\alpha\,c)}{-3\,a\,\beta+2\,a\,\gamma+3\,\alpha\,b-2\,\alpha\,c-b\,\gamma+\beta\,c}-\frac{\gamma\,(a\,\beta-\alpha\,b)}{-3\,a\,\beta+2\,a\,\gamma+3\,\alpha\,b-2\,\alpha\,c-b\,\gamma+\beta\,c},$$

$$\frac{\alpha\,(3\,b-2\,c)}{-3\,a\,\beta+2\,a\,\gamma+3\,\alpha\,b-2\,\alpha\,c-b\,\gamma+\beta\,c}-\frac{\beta\,(3\,a-c)}{-3\,a\,\beta+2\,a\,\gamma+3\,\alpha\,b-2\,\alpha\,c-b\,\gamma+\beta\,c}$$

$$+\frac{\gamma(2a-b)}{-3a\beta+2a\gamma+3\alpha b-2\alpha c-b\gamma+\beta c}\Bigg]\Bigg]$$

```
> simplify(%);
```

$$\begin{bmatrix}1 & 0 & 0\\0 & 1 & 0\\0 & 0 & 1\end{bmatrix} \tag{5.26}$$

Abaixo usamos o comando matriz(m,n,[lista]).

```
> M2:=matrix( 3 ,3 ,[1,4,4,-3,7,0,0,2,7]);
```

$$M2 := \begin{bmatrix}1 & 4 & 4\\-3 & 7 & 0\\0 & 2 & 7\end{bmatrix} \tag{5.27}$$

O comando abaixo reduz a matriz a forma triangular superior

```
> gausselim(M2);
```

$$\begin{bmatrix}1 & 4 & 4\\0 & 2 & 7\\0 & 0 & -\frac{109}{2}\end{bmatrix} \tag{5.28}$$

Aumentando M2 com a identidade.

```
> Id:=diag(1,1,1);
```

$$Id := \begin{bmatrix}1 & 0 & 0\\0 & 1 & 0\\0 & 0 & 1\end{bmatrix} \tag{5.29}$$

```
> M3:=extend(M2, 0, 3, 0 );
```

$$M3 := \begin{bmatrix}1 & 4 & 4 & 0 & 0 & 0\\-3 & 7 & 0 & 0 & 0 & 0\\0 & 2 & 7 & 0 & 0 & 0\end{bmatrix} \tag{5.30}$$

```
> MA:=copyinto( Id, M3 ,1, 4 );
```

$$MA := \begin{bmatrix}1 & 4 & 4 & 1 & 0 & 0\\-3 & 7 & 0 & 0 & 1 & 0\\0 & 2 & 7 & 0 & 0 & 1\end{bmatrix} \tag{5.31}$$

O seguinte comando reduz a matriz a forma escada. Neste exemplo usamos um teorema da Algebra Linear para
determinar a inversa de M2.

```
> gaussjord(MA);
```

(5.32)

$$\begin{bmatrix} 1 & 0 & 0 & \frac{49}{109} & -\frac{20}{109} & -\frac{28}{109} \\ 0 & 1 & 0 & \frac{21}{109} & \frac{7}{109} & -\frac{12}{109} \\ 0 & 0 & 1 & -\frac{6}{109} & -\frac{2}{109} & \frac{19}{109} \end{bmatrix} \tag{5.32}$$

Qual é a inversa de M2 ?

Agora vamos ver como resolver sistemas lineares no Maple usando o comando "linsolve".

Resolver o sistema de Eqs. lineares

```
> F:=array( [ [-1,2,4], [3 ,2 ,1], [6 ,0 ,-3] ] );
```

$$F := \begin{bmatrix} -1 & 2 & 4 \\ 3 & 2 & 1 \\ 6 & 0 & -3 \end{bmatrix} \tag{5.33}$$

```
> B:=array( [ [1], [2], [3] ] );
```

$$B := \begin{bmatrix} 1 \\ 2 \\ 3 \end{bmatrix} \tag{5.34}$$

```
> X:=linsolve(F,B);
```

$$X := \begin{bmatrix} 1 \\ -1 \\ 1 \end{bmatrix} \tag{5.35}$$

Testando a solução.

```
> multiply(F,X);
```

$$\begin{bmatrix} 1 \\ 2 \\ 3 \end{bmatrix} \tag{5.36}$$

Polinômio Característico, autovalores, etc...

```
> C:=matrix([[2,5],[4,8]]);
```

$$C := \begin{bmatrix} 2 & 5 \\ 4 & 8 \end{bmatrix} \tag{5.37}$$

```
> p:=charpoly(C, lambda);  # Polinomio característico
```

$$p := \lambda^2 - 10\,\lambda - 4 \tag{5.38}$$

```
> solve(p, lambda);
```

$$5 + \sqrt{29},\, 5 - \sqrt{29} \tag{5.39}$$

```
> eigenvals(C);
```

$$5 + \sqrt{29},\, 5 - \sqrt{29} \tag{5.40}$$

```
> ### WARNING: note that `I` is no longer of type `radical`
  eigenvects(C, radical);# na ordem: autovalor,
  multiplicidade, autovetor
```

$$\left[5+\sqrt{29}, 1, \left\{\left[\begin{array}{cc} 1 & \frac{3}{5}+\frac{1}{5}\sqrt{29} \end{array}\right]\right\}\right], \left[5-\sqrt{29}, 1, \left\{\left[\begin{array}{cc} 1 & \frac{3}{5}-\frac{1}{5}\sqrt{29} \end{array}\right]\right\}\right] \tag{5.41}$$

```
> L:=matrix([[1,1,0,0,0],[-1,1,0,0,0],[0,0,2,1,0],[0,0,0,1,
  -1],[0,0,0,2,4]]);
```

$$L := \left[\begin{array}{ccccc} 1 & 1 & 0 & 0 & 0 \\ -1 & 1 & 0 & 0 & 0 \\ 0 & 0 & 2 & 1 & 0 \\ 0 & 0 & 0 & 1 & -1 \\ 0 & 0 & 0 & 2 & 4 \end{array}\right] \tag{5.42}$$

```
> solve(pl, lambda);
> eigenvals(L);
```

$$3, 1-I, 1+I, 2, 2 \tag{5.43}$$

```
> ### WARNING: note that `I` is no longer of type `radical`
  eigenvects(L, radical);
```

$$\left[1+I, 1, \left\{\left[\begin{array}{ccccc} 1 & I & 0 & 0 & 0 \end{array}\right]\right\}\right], \left[1-I, 1, \left\{\left[\begin{array}{ccccc} 1 & -I & 0 & 0 & 0 \end{array}\right]\right\}\right], \left[3, 1, \left\{\left[\begin{array}{ccccc} 0 & 0 & 1 & 1 & -2 \end{array}\right]\right\}\right], \left[2, 2, \left\{\left[\begin{array}{ccccc} 0 & 0 & 1 & 0 & 0 \end{array}\right]\right\}\right] \tag{5.44}$$

```
> jordan(L);
```

$$\left[\begin{array}{ccccc} 3 & 0 & 0 & 0 & 0 \\ 0 & 1-I & 0 & 0 & 0 \\ 0 & 0 & 1+I & 0 & 0 \\ 0 & 0 & 0 & 2 & 1 \\ 0 & 0 & 0 & 0 & 2 \end{array}\right] \tag{5.45}$$

Vamos determinar P tal que :

```
> J := jordan(L, 'P');
> print(P);
```

$$J := \left[\begin{array}{ccccc} 3 & 0 & 0 & 0 & 0 \\ 0 & 1-I & 0 & 0 & 0 \\ 0 & 0 & 1+I & 0 & 0 \\ 0 & 0 & 0 & 2 & 1 \\ 0 & 0 & 0 & 0 & 2 \end{array}\right]$$

$$\left[\begin{array}{ccccc} 0 & \frac{1}{2} & \frac{1}{2} & 0 & 0 \\ 0 & -\frac{1}{2}I & \frac{1}{2}I & 0 & 0 \\ -1 & 0 & 0 & 2 & 1 \\ -1 & 0 & 0 & 0 & 2 \\ 2 & 0 & 0 & 0 & -2 \end{array}\right] \tag{5.46}$$

Voltando a coisas básicas

```
> vandermonde([1,0,2]);
```

$$\begin{bmatrix} 1 & 1 & 1 \\ 1 & 0 & 0 \\ 1 & 2 & 4 \end{bmatrix} \tag{5.47}$$

```
> VV:=vandermonde( [x,y,z] );
```

$$VV := \begin{bmatrix} 1 & x & x^2 \\ 1 & y & y^2 \\ 1 & z & z^2 \end{bmatrix} \tag{5.48}$$

```
> det(VV);
```

$$-x^2 y + x^2 z + x y^2 - x z^2 - y^2 z + y z^2 \tag{5.49}$$

```
> factor(%);
```

$$-(y-z)(x-z)(x-y) \tag{5.50}$$

Portanto se x,y,z,... são todos distintos, a matriz de vandermonde é sempre inversível.

```
> vandermonde([1,2,3]);
```

$$\begin{bmatrix} 1 & 1 & 1 \\ 1 & 2 & 4 \\ 1 & 3 & 9 \end{bmatrix} \tag{5.51}$$

```
> J:=matrix([ [1,1,1],[1,2,4], [1,3,9] ]);
```

$$J := \begin{bmatrix} 1 & 1 & 1 \\ 1 & 2 & 4 \\ 1 & 3 & 9 \end{bmatrix} \tag{5.52}$$

```
> jordan(J);
```

$$\Bigg[\bigg[\left(35+\mathrm{I}\sqrt{106}\right)^{1/3}+\frac{11}{\left(35+\mathrm{I}\sqrt{106}\right)^{1/3}}+4, 0, 0\bigg], \\ \bigg[0, -\frac{1}{2}\left(35+\mathrm{I}\sqrt{106}\right)^{1/3}-\frac{11}{2\left(35+\mathrm{I}\sqrt{106}\right)^{1/3}}+4-\frac{1}{2}\,\mathrm{I}\sqrt{3}\left(\left(35+\mathrm{I}\sqrt{106}\right)^{1/3}-\frac{11}{\left(35+\mathrm{I}\sqrt{106}\right)^{1/3}}\right), 0\bigg], \\ \bigg[0, 0, -\frac{1}{2}\left(35+\mathrm{I}\sqrt{106}\right)^{1/3}-\frac{11}{2\left(35+\mathrm{I}\sqrt{106}\right)^{1/3}}+4+\frac{1}{2}\,\mathrm{I}\sqrt{3}\left(\left(35+\mathrm{I}\sqrt{106}\right)^{1/3}-\frac{11}{\left(35+\mathrm{I}\sqrt{106}\right)^{1/3}}\right)\bigg]\Bigg] \tag{5.53}$$

```
> evalf(%);
```

$$[[10.60311024 - 1.\ 10^{-10}\,\mathrm{I}, 0., 0.], \\ [0., 1.245437886 + 9.660254040\ 10^{-10}\,\mathrm{I}, 0.], \tag{5.54}$$

$[0., 0., 0.1514518720 - 7.660254040\ 10^{-10}\ I]]$

Simplex

Problemas de Programação linear são resolvido usando-se o método simplex.
Para isto devemos chamar o pacote "simplex".
Basicamente a programação linear tem por objetivo maximizar ou minimzar um funcional linear sujeito a restrições dadas por desigualdades lineares. Veja um exemplo
Maximizar $-x + y + 2z$, sujeito as seguintes restrições
$3x + 4y - 3z \le 23$
$5x - 4y - 3z \le 10$
$7x + 4y + 11z \le 30$

com x, y e z não negativos.

A teoria de programação linear, diz que o ponto ótimo ocorre em um dos vértices da região determinada pelas desigualdades. O Método simplex, percorre os vértices, procurando o ponto que otimiza a função objetivo.

```
> with(simplex):
> restr1 := {3*x+4*y-3*z <= 23, 5*x-4*y-3*z <= 10,7*x+4*y+11*z <= 30};
```

$$restr1 := \{3x + 4y - 3z \le 23, 5x - 4y - 3z \le 10, 7x + 4y + 11z \le 30\} \quad (5.1.1)$$

```
> obj1 := -x + y + 2*z;
```

$$obj1 := -x + y + 2z \quad (5.1.2)$$

```
> maximize(obj1,restr1 union {x>=0,y>=0,z>=0});
```

$$\left\{x = 0, y = \frac{49}{8}, z = \frac{1}{2}\right\} \quad (5.1.3)$$

```
> maximize(obj1,restr1,NONNEGATIVE);
```

$$\left\{x = 0, y = \frac{49}{8}, z = \frac{1}{2}\right\} \quad (5.1.4)$$

```
> restr2 :={3*x+4*y <= 24,5*x-4*y <= 12}:obj2 := -x + y :
> minimize(obj2,restr2,NONNEGATIVE);
```

$$\left\{x = \frac{12}{5}, y = 0\right\} \quad (5.1.5)$$

```
> maximize(obj2,restr2,NONNEGATIVE);
```

$$\{x = 0, y = 6\} \quad (5.1.6)$$

Veja a região determinada pelas desigualdades restr2.

```
> plot({(24-3*x)/4 ,(-12+5*x)/4},x=-1..10);
```




```
> with(simplex):
> display({x+3*y+z<=0,w-2*y-z<=2});
```

$$\begin{bmatrix} 0 & 1 & 3 & 1 \\ 1 & 0 & -2 & -1 \end{bmatrix} \begin{bmatrix} w \\ x \\ y \\ z \end{bmatrix} \le \begin{bmatrix} 0 \\ 2 \end{bmatrix} \tag{5.1.7}$$

5. **Quinta parte** - **Procedimentos**

No Maple os programas são feitos em forma de procedimentos. Um procedimento é na verdade uma função que transforma os argumentos inseridos em outros argumentos, conforme foi programado. A estrutura de um procedimento é a seguinte

```
Nomeproced:=proc(argumentos)
        local (variáveis locais)
        instruções a serem executadas
        end;
```

Num procedimento podem ser necessários comandos de

repetição e iteração
seleção

Vamos falar rapidamente sobre eles.

Certas situações exigem que uma instrução seja repetida várias vezes. Para isto utilizamos o comando **"for"**. A utilização deste comando segue o esquema
(for-do-od)

Por exemplo:
for j from inicio by passo to fim
do expressões dependentes de j
od

A sequência **"do-od"** também funciona em outros comandos.
Um outro comando especial para fazer recorrências é o **"while"** que em português significa enquanto. O esquema é o seguinte:
while k satisfaz-condição do
instruções envolendo k
od

Os comandos de seleção **"if"** ou desvio são utilizados para se executar uma instrução dentre as várias possiveis, condicionadas a uma proposição que pode ser falsa ou verdadeira. O esquema é o seguinte:
if condição verdadeira then
instruções a executar
else outras instruçoes a executar
fi

Os camandos de saída de dados são **"print"**e o **"lprint"**.
A diferença entre eles vai ficar clara nos exemplos.

```
> restart;
```

Exemplo 1. Vamos ver um procedimento para determinar se um dado número é par.

```
> pp:=proc(p) if irem(p,2)=0 then print(p,`é par`) else print
  (p,`é impar`) fi end;
```

$pp := \mathbf{proc}(p)$ **(6.1)**
$\quad \mathbf{if}\ irem(p, 2) = 0\ \mathbf{then}\ print(p, \text{`é par`})\ \mathbf{else}\ print(p, \text{`é impar`})\ \mathbf{end\ if}$
$\mathbf{end\ proc}$

```
> pp(-3);
```

$$-3, \textit{é impar} \qquad \textbf{(6.2)}$$

```
> pp(6);
```

$$6, \textit{é par} \qquad \textbf{(6.3)}$$

```
> pp(9);
```

$$9, \textit{é impar} \qquad \textbf{(6.4)}$$

```
> pp(2.4);# deve acusar erro.
Error, (in pp) invalid input: irem received 2.4, which is 
not valid for its 1st argument, m
```

```
> pp(192837465);
```

$$192837465, \textit{é impar} \qquad \textbf{(6.5)}$$

```
> restart;
```

Exemplo 2. Escrever os primeiros p números naturais múltiplos de 7.

```
> m7:=proc(p) local k; for k from 1 to p do print(  k*7  ); od
  end;
```

$m7 := \mathbf{proc}(p)\ \mathbf{local}\ k;\ \mathbf{for}\ k\ \mathbf{to}\ p\ \mathbf{do}\ print(7*k)\ \mathbf{end\ do}\ \mathbf{end\ proc}$ **(6.6)**

```
> m7(5);
```

7
14
21
28
35 **(6.7)**

```
> m7(50);
```

7
14
21
28
35
42
49
56
63
70
77
84
91
98
105
112
119
126
133
140
147
154
161
168
175
182
189
196
203
210
217
224
231
238
245
252
259
266
273
280

287
294
301
308
315
322
329
336
343
350 **(6.8)**

```
> restart;
```

Exemplo3. O número de quadrados perfeitos menores ou iguais a 112.

```
> k:=0: # k é o contador.
> while k^2<=112 do k:= k+1 od:
> print(` o numero de quadrados menores do que ou iguais 112
  `, k);
```

o numero de quadrados menores do que ou iguais 112 , 11 **(6.9)**

```
> restart;
```

Exemplo 4. As raizes de uma equação do segundo grau

```
> R:=proc(a,b,c) if b^2-4*a*c<0 then print(`as raizes sao
  complexas` -b/(2*a)+sqrt(b^2-4*a*c)/(2*a) , -b/(2*a)-sqrt
  (b^2-4*a*c)/(2*a)) else print(` as raizes sao`-b/(2*a)+sqrt
  (b^2-4*a*c)/(2*a), -b/(2*a)-sqrt(b^2-4*a*c)/(2*a)) fi end:
> R(1,0,1);## solução de x^2+1=0
```

$\mathit{as\ raizes\ sao\ complexas} + \mathrm{I}, -\mathrm{I}$ **(6.10)**

```
> R(-1,5,6);### solução de -1x^2+5x+6=0
```

$\mathit{as\ raizes\ sao} - 1, 6$ **(6.11)**

```
> R(1,-5,6);### solução de 1x^2-5x+6=0
```

$\mathit{as\ raizes\ sao} + 3, 2$ **(6.12)**

Exemplo 5 . ponto fixo.
Este procedimento determina uma aproximação do ponto fixo de uma aplicação usando sucessivas iterações
.

Você entra com a aplicação, o chute inicial e o número de iterações.

```
> pfixo1:=proc(f,chute,n)
local x,k;
x[0]:=evalf( chute );
for k from 1 to n do
x[k]:=evalf( f(x[k-1]) );
od;
if abs(x[n]-x[n-1])-abs(x[n-1]-x[n-2])> 0 then ERROR(`a seq
diverge`)
fi;
print(evalf(x[n]));
end:
```

Testando: um pontofixo de $f(x) = \sqrt{x}$ é x=1.

```
> pfixo1(sqrt, 0.9 ,30);
```
$$0.9999999999 \quad (6.13)$$

```
> f:=x->x^3-x;
```
$$f := x \rightarrow x^3 - x \quad (6.14)$$

```
> pfixo1(f,.5,20);
```
$$0.1440914959 \quad (6.15)$$

Problema 1. Use o comando print ou lprint para estabelecer algum diálogo com o usuário.

Problema 2. Incluir no programa "pfixo1" um comando de forma a avisar o usuário quando as iterações divergem.

Exemplo 6. **Cálculo de raízes via bissecção.**
Aqui você entra com a função, com os extremos do intervalo, o numero n de bissecções.

```
bissec1:=proc(f,a,b,n)
   # argumentos: f=funcao
   #             a,b  extremos do intervalo [a,b]
   #             n=numero de bisseccoes
   local aa, bb, c, k:  # variaveis locais
   aa:=a;
   bb:=b;
if aa-bb=0 then ERROR(`a`,aa=`b`,bb);
fi;
for k from 1 to n do
   c[k]:=evalf( (aa+bb)/2 );
   if evalf( f(aa)*f(c[k]) )>0 then aa:=c[k]
elif evalf( f(aa)*f(c[k]) )=0 then ERROR(`f(aa)*f(bb)=0`)
   else bb:=c[k]
   fi
od;
   print(` Raiz aproximada após`,n,`bissecções:` );
   print( evalf(c[n]) );
end:
```

Testando o nosso procedimento.

```
> ff:= x -> x^3+3*x-10;
```
$$ff := x \rightarrow x^3 + 3\,x - 10 \quad (6.16)$$

```
> bissec1(ff, 1, 4, 30);
```
Raiz aproximada após, 30, *bissecções:*
$$1.698885492 \quad (6.17)$$

```
> bissec1(sin, 2, 4, 15);
```
Raiz aproximada após, 15, *bissecções:*
$$3.141540527 \quad (6.18)$$

Exemplo 7. Usamos a sequência $x_k = \dfrac{x_{k-1} + \dfrac{N}{x_{k-1}}}{2}$ para o cálculo de uma aproximação para N.

```
> seqq:=proc(n,chute,m) local x,k;  x[0]:= chute;  for k from 1 to m do  x[k]:= evalf((1/2)*(x[k-1]+n/x[k-1])); od; print(evalf(x[m])); end:
```

```
> seqq(3,1,20); # raiz quadrada de 3 apos 20 iteracoes.
```

$$1.732050808 \tag{6.19}$$

```
> seqq(5,1,10); # raiz quadrada de 5 apos 10 iteracoes.
```

$$2.236067977 \tag{6.20}$$

```
> Digits:= 30:# para pedir uma aproximaçao com 30 digitos
> seqq(2,1,20); # raiz quadrada de 2 apos 20 iteracoes.
```

$$1.41421356237309504880168872421 \tag{6.21}$$

```
> seqq(5,1,100); # raiz quadrada de 5 apos 100 iteracoes.
```

$$2.23606797749978969640917366873 \tag{6.22}$$

6. **Sexta parte - Equações diferenciais ordinárias**

O comando básico do Maple para resolver equações diferenciais ordinarias é o "dsolve".
A sintaxe do dsolve é dsolve("oque", "como");

O "oque" refere-se a EDO (ou sistema de edo's) junto com condições iniciais. O "como" especifica qual rotina do Maple vai ser utilizada.

É útil dar nomes a todas as equações e condições iniciais quando usamos o dsolve.

Consulte o help para saber sobre EDO. Tecle `???odesolve;`

Consulte o help para saber sobre EDP. Tecle `???pdesolve;`

Exemplos:
Para aprender a usar o help: entre com o comando `with(DEtools);`

```
> restart:
> eq:=diff(y(x),x)=x*y(x);
```

$$eq := \frac{d}{dx}\, y(x) = x\, y(x) \tag{7.1}$$

```
> init:=y(2)=1;
```

$$init := y(2) = 1 \tag{7.2}$$

Agora, **eq** é o nome da edo que queremos resolver e **init** é o nome das condições iniciais. É importante usar y(x) -- isto indica que tomamos y como uma variavel dependente e x como variavel independente.
Para resolver sem a condição inicial, isto é achar a solução geral, usamos `dsolve`

```
> dsolve(eq,y(x));
```

$$y(x) = \_C1\, e^{\frac{1}{2}x^2} \tag{7.3}$$

Note que _C1 é a constante arbitraria produzida pelo maple.

Para resolver o PVI devemos agrupar a EDO junto com a condicao inicial entre chaves.

```
> dsolve({eq,init},y(x));
```

$$y(x) = \frac{e^{\frac{1}{2}x^2}}{e^2} \tag{7.4}$$

As vezes Maple dá uma resposta implicita, assim é comum pedir que o Maple explicite a resposta em função da variavel dependente. Exemplo:

```
> dsolve(diff(y(x),x)=y(x)^2, y(x));
```

$$y(x) = \frac{1}{-x + \_C1} \tag{7.5}$$

Usamos o comando explicit como opcao em dsolve:

```
> dsolve(diff(y(x),x)=y(x)^2, y(x), explicit);
```

$$y(x) = \frac{1}{-x + \_C1} \tag{7.6}$$

Uso mais avançado de dsolve:

Existem 4 maneiras de usar o dsolve para situações além das edos de primeira ordem e pvi simples.

1. Equações de ordem superior
2. Sistemas de EDO's
3. Soluções numericas
4. Usando séries de potências para obter soluções de edo's

1. Equações de ordem superior : edo's de ordem 2 ou superior podem ser resolvidas usando o dsolve. As derivadas de ordem superior são escritas como no exemplo.

```
> diff(f(x),x$3);
```

$$\frac{d^3}{dx^3} f(x) \tag{7.7}$$

Uma EDO de ordem 2:

```
> eqn2:=diff(y(x),x$2)+3*diff(y(x),x)+2*y(x)=exp(x);
```

$$eqn2 := \frac{d^2}{dx^2} y(x) + 3 \left(\frac{d}{dx} y(x) \right) + 2\, y(x) = e^x \tag{7.8}$$

A solução geral é:

```
> dsolve(eqn2,y(x));
```

$$y(x) = \frac{1}{6} e^x - e^{-2x}\_C1 + e^{-x}\_C2 \tag{7.9}$$

Para este problema vamos especificar condições de y em x=1:

```
> inits:=y(1)=2, D(y)(1)=4;
```

$$inits := y(1) = 2, D(y)(1) = 4 \tag{7.10}$$

```
> dsolve({eqn2,inits},y(x));
```

$$y(x) = \frac{1}{6}\, e^{x} + \frac{1}{3}\, \frac{e^{-2x}\,(e-18)}{e^{-2}} - \frac{1}{2}\, \frac{e^{-x}\,(e-16)}{e^{-1}} \tag{7.11}$$

2. **Sistemas de EDO's:** Vamos ver um exemplo.

```
> eqns:=diff(y(x),x)+diff(z(x),x)=x, diff(y(x),x)-2*diff(z(x),
  x)=x^2;
```

$$eqns := \frac{d}{dx}\, y(x) + \frac{d}{dx}\, z(x) = x,\ \frac{d}{dx}\, y(x) - 2\left(\frac{d}{dx}\, z(x)\right) = x^{2} \tag{7.12}$$

```
> inits:= y(0)=1, z(0)=2;
```

$$inits := y(0) = 1, z(0) = 2 \tag{7.13}$$

```
> dsolve({eqns,inits},{y(x),z(x)});
```

$$\left\{y(x) = \frac{1}{9}\, x^{3} + \frac{1}{3}\, x^{2} + 1, z(x) = -\frac{1}{9}\, x^{3} + \frac{1}{6}\, x^{2} + 2\right\} \tag{7.14}$$

```
> restart:# vamos zerar a memoria do maple
```

3. **Soluções Numéricas:** Em geral é impossivel obter explicitamente a soluçao de uma EDO, neste caso usamos um método numérico para aproximar a solução. O Maple faz isto, basta usar a opcao numeric em dsolve, como no exemplo:

```
> eqn:=diff(y(x),x)+exp(y(x))*x^3=2*sin(x); init:=y(0)=2;
```

$$eqn := \frac{d}{dx}\, y(x) + e^{y(x)}\, x^{3} = 2 \sin(x)$$

$$init := y(0) = 2 \tag{7.15}$$

```
> F:=dsolve({eqn,init},y(x),numeric);
```

$$F := \mathbf{proc}(x\_rkf45)\ \ldots\ \mathbf{end\ proc} \tag{7.16}$$

O valor de F em x = 2 é

```
> F(2);
```

$$[x = 2., y(x) = -0.781597136489398] \tag{7.17}$$

```
>
```

Em geral é util plotar numericamente a solução obtida de uma EDO. Para fazer isto usamos o comando **odeplot** ao resultado (F neste caso) de dsolve(....,numeric) . Para usar odeplot precisamos de "chamar" o plots .

```
> with(plots, odeplot);
```

$$[odeplot] \tag{7.18}$$

```
> odeplot(F, [x, y(x)],-2..2);
```



>

4. **Usando series de potências**. Exemplo: resolver y'+x*y=0, y(0)=1 por series .

```
> dsolve({diff(y(x), x) + x * y(x) = 0, y(0) = 1}, y(x), series);
```

$$y(x) = 1 - \frac{1}{2}x^2 + \frac{1}{8}x^4 + O(x^6) \quad \textbf{(7.19)}$$

Um pouco de EDP

```
> restart:
> with(PDEtools);
```

[*CanonicalCoordinates, ChangeSymmetry, CharacteristicQ, CharacteristicQInvariants, ConservedCurrentTest, ConservedCurrents, ConsistencyTest, D_Dx, DeterminingPDE, Eta_k, Euler, FromJet, FunctionFieldSolutions, InfinitesimalGenerator, Infinitesimals, IntegratingFactorTest, IntegratingFactors, InvariantEquation, InvariantSolutions, InvariantTransformation, Invariants, Laplace, Library, PDEplot, PolynomialSolutions, ReducedForm, SimilaritySolutions, SimilarityTransformation, Solve, SymmetryCommutator, SymmetryGauge, SymmetrySolutions, SymmetryTest, SymmetryTransformation, TWSolutions,* **(7.1.1)**

ToJet, build, casesplit, charstrip, dchange, dcoeffs, declare, diff_table, difforder, dpolyform, dsubs, mapde, separability, splitstrip, splitsys, undeclare]

```
> PDE  := x^2*diff(f(x,y),y)-y^2*diff(f(x,y),x) = 0;
> sol:=pdsolve(PDE);
```

$$PDE := x^2 \left(\frac{\partial}{\partial y} f(x, y) \right) - y^2 \left(\frac{\partial}{\partial x} f(x, y) \right) = 0$$

$$sol := f(x, y) = \_F1\left(x^3 + y^3\right) \tag{7.1.2}$$

```
> pdetest(sol,PDE);### vamos testar??
```

$$0 \tag{7.1.3}$$

```
> eqp:=diff(f(x,y),y)+3*diff(f(x,y),x)+2*f(x,y)=1;
```

$$eqp := \frac{\partial}{\partial y} f(x, y) + 3 \left(\frac{\partial}{\partial x} f(x, y) \right) + 2 f(x, y) = 1 \tag{7.1.4}$$

A solução geral é:

```
> sol2:=pdsolve(eqp,f(x,y));
```

$$sol2 := f(x, y) = \frac{1}{2} + e^{-\frac{2}{3} x} \_F1\left(-\frac{1}{3} x + y \right) \tag{7.1.5}$$

```
> pdetest(sol2,eqp);
```

$$0 \tag{7.1.6}$$

```
>
```

7. Sétima Parte - Mudança de variáveis

Quando fazemos mudança de variáveis, o determinante da matriz Jacobiana desta tranformação, aparece em várias situações. Uma delas é no cálculo de integrais: você muda a variável para simplificar a integração, mas precisar incluir o módulo do determinante da matriz Jacobiana da mudança de variáveis na sua nova integral.

O Maple calcula a matriz Jacobiana e o seu determinate. Vamos ver exemplos.

```
> with(linalg):
```

Exemplo 1: a mudança de coordenas polares   e tem jacobiano dada por

```
> J:=jacobian([r*cos(theta),r*sin(theta)],[r,theta]);
```

$$J := \begin{bmatrix} \cos(\theta) & -r\sin(\theta) \\ \sin(\theta) & r\cos(\theta) \end{bmatrix} \tag{8.1}$$

```
> `det(J)`:=simplify(det(J));
```

$$det(J) := r \tag{8.2}$$

Exemplo 2: a mudança de coordenas esfericas tem jacobiano dada por

```
> J:=jacobian([rho*cos(theta)*sin(phi),
              rho*sin(theta)*sin(phi),
              rho*cos(phi)],
             [rho,theta,phi]);
```

(8.3)

$$J := \begin{bmatrix} \cos(\theta)\sin(\phi) & -\rho\sin(\theta)\sin(\phi) & \rho\cos(\theta)\cos(\phi) \\ \sin(\theta)\sin(\phi) & \rho\cos(\theta)\sin(\phi) & \rho\sin(\theta)\cos(\phi) \\ \cos(\phi) & 0 & -\rho\sin(\phi) \end{bmatrix} \quad \textbf{(8.3)}$$

```
> `det(J)`:=simplify(det(J));
```

$$det(J) := -\sin(\phi)\,\rho^2 \quad \textbf{(8.4)}$$

Exemplo 3

```
> J:=jacobian([u+v,u-v],[u,v]);
```

$$J := \begin{bmatrix} 1 & 1 \\ 1 & -1 \end{bmatrix} \quad \textbf{(8.5)}$$

```
> `det(J)`:=simplify(det(J));
```

$$det(J) := -2 \quad \textbf{(8.6)}$$

Exemplo 4

```
> J:=jacobian([u^2-v+w,u-v-w,u+v+w],[u,v,w]);
```

$$J := \begin{bmatrix} 2u & -1 & 1 \\ 1 & -1 & -1 \\ 1 & 1 & 1 \end{bmatrix} \quad \textbf{(8.7)}$$

```
> `det(J)`:=simplify(det(J));
```

$$det(J) := 4 \quad \textbf{(8.8)}$$

Exemplo 5

```
> JacobianoC := jacobian([r*cos(theta),r*sin(theta),z],[r,
  theta,z]);
```

$$JacobianoC := \begin{bmatrix} \cos(\theta) & -r\sin(\theta) & 0 \\ \sin(\theta) & r\cos(\theta) & 0 \\ 0 & 0 & 1 \end{bmatrix} \quad \textbf{(8.9)}$$

```
> DetjacobianoC := simplify(det(JacobianoC));
```

$$DetjacobianoC := r \quad \textbf{(8.10)}$$

Uma aplicação

Como exemplo considere o sólido W abaixo do plano z=1 e dentro do cone circular reto superior z=

Vamos plotar seu gráfico.

```
> with(plots):
```

```
> cylinderplot(z,theta = 0..2*Pi,z=0..1);
```



O volume de W usando coordenadas cilindrinas é

```
> Volume := Int(Int(Int(r,z=r..1),r=0..1),theta=0..2*Pi)=int
  (int(int(r,z=r..1),r=0..1),theta=0..2*Pi);
```

$$Volume := \int_0^{2\pi}\int_0^1\int_r^1 r \,\mathrm{d}z\,\mathrm{d}r\,\mathrm{d}\theta = \frac{1}{3}\pi \qquad \textbf{(8.11)}$$

8 . Oitava parte -Integrais múltiplas

> $Int(x+y, x=0..2, y=1..3) = int(x+y, x=0..2, y=1..3);$

$$\int_1^3\int_0^2 (x+y)\,\mathrm{d}x\,\mathrm{d}y = 12 \qquad \textbf{(9.1)}$$

> $Int(x+y+z, x=0..2, y=1..3, z=2..5) = int(x+y+z, x=0..2, y=1..3, z=2..5);$

$$\int_2^5\int_1^3\int_0^2 (x+y+z)\,\mathrm{d}x\,\mathrm{d}y\,\mathrm{d}z = 78 \qquad \textbf{(9.2)}$$

> $Int(\exp(x+y), x=0..2, y=1..3) = int(\exp(x+y), x=0..2, y=1..3);$

$$\int_1^3\int_0^2 e^{x+y}\,\mathrm{d}x\,\mathrm{d}y = e - 2e^3 + e^5 \qquad \textbf{(9.3)}$$

> $Int(\cos(x+y), x=0..2, y=1..3) = int(\cos(x+y), x=0..2, y=1..3);$

$$\int_1^3\int_0^2 \cos(x+y)\,dx\,dy = -\cos(1)+2\cos(3)-\cos(5) \quad (9.4)$$

> $int\left(\frac{x}{x^3+1}, x=\frac{3}{4}..\frac{5}{4}, numeric\right);$

$$0.2459707569 \quad (9.5)$$

> $int\left(\frac{1}{x^2+1}, x=\frac{3}{4}..\frac{5}{4}, numeric=false\right);$

$$-\arctan\left(\frac{3}{4}\right)+\arctan\left(\frac{5}{4}\right) \quad (9.6)$$

9. Nona parte - Campos vetoriais convervativos

Vamos ilustrar o uso do Maple no cálculo de potenciais de campos conservativos. Primeiramente relembramos algumas definições e resultados básicos.

1. Introdução

Se : --> está definido sobre a região do e existe g: --> R tal que = F, então dizemos que é um campo conservativo e que é um potencial para .

No cálculo de gradientes, vamos usar um pacote de Álgebra Linear. Vamos relembrar a definição de gradiente.

> restart:

```
> with(linalg):
> 'grad(g,[x,y,z])'=[Diff(g,x),Diff(g,y),Diff(g,z)];
```

$$linalg\text{:-}grad(g, [x, y, z]) = \left[\frac{\partial}{\partial x} g, \frac{\partial}{\partial y} g, \frac{\partial}{\partial z} g\right] \quad (10.1.1)$$

Exemplo 1: Seja . Este é um campo conservativo e é um potencial para .

```
> grad(x*y*z, [x,y,z]);
```

$$\left[\, y\,z \quad x\,z \quad x\,y \,\right] \quad (10.1.2)$$

Se F é um campo conservativo e c descreve uma curva fechada , então a integral de F ao longo de c é zero, onde a integral é tomada sobre a curva fechada.

Exemplo 2: A função F(x,y,z) = [y, -x, 0] não é um campo conservativo, pois tomando a curva como sendo um círculo de raio 1 no plano XY, vemos que

```
> Int(dotprod([sin(t),-cos(t)],[diff(cos(t),t),diff(sin(t),
  t)]),t=0..2*Pi)
  = int(dotprod([sin(t),-cos(t)],[diff(cos(t),t),diff(sin
  (t),t)]),t=0..2*Pi);
```

$$\int_0^{2\pi} \left(-\sin(t)\sin(\overline{t})-\cos(t)\cos(\overline{t})\right)dt = -2\pi \quad (10.1.3)$$

2. Como reconhecer se F é conservativo?

Teorema 1 : Seja F contínuo em uma região Ω do espaço. São equivalentes:
(a) é conservativo.
(b) A integral de linha de F sobre um caminho é independente do caminho.

Este é um ótimo resultado. A parte mais interessante deste teorema é a implicação (b) implica (a).

Exemplo 1 (de novo): Vimos que é conservativo, assim pelo teorema 1, a integral sobre qualquer caminho fechado do é zero. Tomemos como exemplo o circulo no plano XY.

```
> F:=(x,y,z)->[y*z,x*z,x*y];
  x:=t->cos(t); y:=t->sin(t); z:=t->0;
  Int('F(x(t),y(t),z(t))[1]'*'D(x)(t)',t=0..2*Pi)
        + Int('F(x(t),y(t),z(t))[2]'*'D(y)(t)',t=0..2*Pi)
              + Int('F(x(t),y(t),z(t))[3]'*'D(z)(t)',t=0..2*
  Pi) =
  int(F(x(t),y(t),z(t))[1]*D(x)(t),t=0..2*Pi)
        + int(F(x(t),y(t),z(t))[2]*D(y)(t),t=0..2*Pi)
              + int(F(x(t),y(t),z(t))[3]*D(z)(t),t=0..2*Pi);
```

$$F := (x, y, z) \to [y\,z, x\,z, y\,x]$$
$$x := t \to \cos(t)$$
$$y := t \to \sin(t)$$
$$z := t \to 0$$
$$\int_0^{2\pi} F(x(t), y(t), z(t))_1 \, \mathrm{D}(x)(t)\, \mathrm{d}t + \int_0^{2\pi} F(x(t), y(t), z(t))_2 \, \mathrm{D}(y)(t)\, \mathrm{d}t + \int_0^{2\pi} F(x(t), y(t), z(t))_3 \, \mathrm{D}(z)(t)\, \mathrm{d}t = 0 \quad \textbf{(10.2.1)}$$

Exemplo 2 (de novo): Vimos que a integral sobre qualquer curva fechada não é necessariamente zero para este exemplo. Vamos ver, de outro modo, que não existe g tal que

$$\frac{\partial}{\partial x} g(x, y) = y \text{ and } \frac{\partial}{\partial y} g(x, y) = -x.$$

Se existisse uma tal g, então as derivadas parciais mistas de segunda ordem deveriam ser iguais.

3. **Como reconhecer se é conservativo?**

Teorema 2: Seja F(x,y,z)=[P(x,y,z),Q(x,y,z), R(x,y,z)] e suponha que P , Q, e R, juntos com suas primeiras derivadas são continuas em uma região simples Ω . São equivalentes:
(a) F é um campo conservativo sobre ,
(b) $\frac{\partial}{\partial y} P = \frac{\partial}{\partial x} Q$, $\frac{\partial}{\partial z} P = \frac{\partial}{\partial x} R$, e $\frac{\partial}{\partial z} Q = \frac{\partial}{\partial y} R$ em Ω.

Exemplo 1: Vamos verificar as derivadas parciais com o campo F(x,y,z)=(yz,xz,xy) .

```
> x:='x': y:='y': z:='z':
  F:=(x,y,z)->[y*z,x*z,x*y];
  diff(F(x,y,z)[1],y),diff(F(x,y,z)[2],x);
```

```
diff(F(x,y,z)[1],z),diff(F(x,y,z)[3],x);
diff(F(x,y,z)[3],y),diff(F(x,y,z)[2],z);
```

$$F := (x, y, z) \rightarrow [y\,z, x\,z, y\,x]$$
$$z, z$$
$$y, y$$
$$x, x \tag{10.3.1}$$

Exemplo 2: nao é conservativo: F(x,y,z)=(y,x,0).

```
> F:=(x,y,z)->[y,-x,0];
  diff(F(x,y,z)[1],y),diff(F(x,y,z)[2],x);
  diff(F(x,y,z)[1],z),diff(F(x,y,z)[3],x);
  diff(F(x,y,z)[3],y),diff(F(x,y,z)[2],z);
```

$$F := (x, y, z) \rightarrow [y, -x, 0]$$
$$1, -1$$
$$0, 0$$
$$0, 0 \tag{10.3.2}$$

4. **Como encontrar o potencial de um campo conservativo?**

É importante saber como determinar o potencial para um campo conservativo.
A técnica é simples e poder ser programada.
O Maple V tem uma rotina para determinar o potencial de um campo. Veja os comandos.
Exemplo 1 (mais uma vez):

```
> F:=(x,y,z)->[y*z,x*z,x*y];
  potential(F(x,y,z),[x,y,z],'h');
  h;
```

$$F := (x, y, z) \rightarrow [y\,z, x\,z, y\,x]$$
$$\mathit{true}$$
$$x\,y\,z \tag{10.4.1}$$

Exemplo 2 (mais uma vez):

```
> F:=(x,y,z)->[y,-x,0];
  potential(F(x,y,z),[x,y,z],'g');
```

$$F := (x, y, z) \rightarrow [y, -x, 0]$$
$$\mathit{false} \tag{10.4.2}$$

Exemplo 3:

```
> F:=(x,y,z)->[exp(y+2*z),x*exp(y+2*z),2*x*exp(y+2*z)];
  potential(F(x,y,z),[x,y,z],'g');
  g;
```

$$F := (x, y, z) \rightarrow [\mathrm{e}^{y+2z}, x\,\mathrm{e}^{y+2z}, 2\,x\,\mathrm{e}^{y+2z}]$$
$$\mathit{true}$$
$$x\,\mathrm{e}^{y+2z} \tag{10.4.3}$$

Exemplo 4:

```
> F:=(x,y,z)->[exp(x)*sin(y)+2*y,exp(x)*cos(y)+2*x-2*y,0];
  potential(F(x,y),[x,y,z],'g');
  g;
  grad(exp(x)*sin(y)+2*x*y-y^2,[x,y,z]);
```

$$F := (x, y, z) \rightarrow [\mathrm{e}^{x}\sin(y) + 2\,y, \mathrm{e}^{x}\cos(y) + 2\,x - 2\,y, 0]$$
$$\mathit{true}$$
$$\mathrm{e}^{x}\sin(y) + 2\,x\,y - y^{2}$$
$$\left[\begin{array}{ccc} \mathrm{e}^{x}\sin(y) + 2\,y & \mathrm{e}^{x}\cos(y) + 2\,x - 2\,y & 0 \end{array}\right] \tag{10.4.4}$$

10. **Décima parte - Equações não lineares**

```
> restart:
```

Determinando soluções de equações não lineares

Vamos investigar os seguintes métodos de determinar as soluções de equações não lineares:

- Investigação pelo gráfico
- Método Iterativo siimples
- Método de Newton-Raphson

EXEMPLO 1 -   Investigação pelo gráfico

Tomemos a função

```
> f:= x-> 2*x - 3 - ln(x);
```

$$f := x \to 2\,x - 3 - \ln(x) \tag{11.1}$$

e vamos plotar o seu gráfico

```
> plot(f(x), x = 0..2);
```



Vemos claramente que existem duas raizes para a equação
neste intervalo [0,2]. Você pode obter uma aproximação destas soluções usando um desenho?
As soluções são aproximadamente e

Agora vamos confirmar estas soluções usando um método de iteração simples. A equação pode ser reescrita como

Chamemos e vamos procurar então um ponto fixo para .

```
> g:=x->(ln(x) + 3)/2;
```

$$g := x \rightarrow \frac{1}{2}\ln(x) + \frac{3}{2} \qquad \textbf{(11.2)}$$

Avaliemos a primeira aproximação em :

```
> evalf(g(1));
```

$$1.500000000 \qquad \textbf{(11.3)}$$

```
> evalf(g(%));
```

$$1.702732554 \qquad \textbf{(11.4)}$$

Repetindo esta iteração com cada aproximação sucessiva, podemos confirmar que é uma aproximação.

```
> for n from 1 to 10 do; evalf(g(%)); od;
```

1.766117173
1.784391725
1.789538793
1.790978965
1.791381189
1.791493468
1.791524806
1.791533552
1.791535993
1.791536674 **(11.5)**

A existencia de uma solução pode ser visto graficamente se plotarmos juntas as duas expressões e : Vemos que os gráficos das duas expressões se encontram.

```
> plot({x,g(x)}, x = 0..2, 0..2, style=LINE);
```



A convergência pode ser ilustrada desenhando segmentos de reta, fazemos isto usando o procedimento abaixo.
O desenho mostra como as aproximações sucessivas convergem para o ponto fixo que é solução de

```
> cobweb:=proc(expr,initial,number,left,right,bottom,top)
  local f,t,i,n,p1,p2; global x1,x2;  f:=expr: x1:=initial:
  n:=number:  x2:=subs(x=x1,f):  t:=(x1,0,x1,x2):     for i
  from 2 to n do     x||(i+1):=subs(x=x||i,f):  t:=t,(x||i,
  x||i,x||i,x||(i+1)):    od:  p1 := plot({x,expr},x=left..
  right,y=bottom..top);
    p2 := plots[pointplot]([t],style=LINE);
    plots[display](p1,p2);end:
> cobweb(g(x),1,20,0,2,0,2);
```



É importante notar que a condição de convergência
|g'(x)| < 1 é satisfeita numa vizinhança de x=1.7915.
Vamos confirmar isto no caso de pertencer a [1,2].

```
> plot(abs(diff(g(x),x)), x= 1..2);
```



Isto é suficiente, mas não necessário para a convergência.

Agora você pode repetir o procedimento para tentar encontrar outras soluções, usando valores diferentes para primeira aproximação. Note que não pode ser encontrado usando g. Tente uma expressão alternativa, por exemplo,

```
> g:= x -> exp(2*x - 3);
```

$$g := x \rightarrow e^{2x-3} \tag{11.6}$$

Usando a primeira aproximação e repetindo os passos para obter como aproximação Novamente a condição $|g'(x)| < 1$ é satisfeita perto do zero.

O processo de iterações sucessivas pode ser ilustrado usando o procedimento anteriormente dado.

```
> cobweb(g(x),1,5,0,1,0,1);
```




```
> plot(abs(diff(g(x),x)), x= 0..1);
```



Maple encontra os zeros numericalmente usando o comando **fsolve**:

```
> fsolve(f(x),x);
```

$$0.05564832190 \tag{11.7}$$

O Maple também precisa conhecer a vizinhança do segundo zero.

```
> fsolve(f(x),x, 0..1);
```

$$0.05564832190 \tag{11.8}$$

EXEMPLO 2: O MÉTODO NEWTON-RAPHSON

Considere a função :

```
> f:= x -> exp(x) - sin(x);
```

$$f := x \rightarrow e^{x} - \sin(x) \tag{11.9}$$

```
> plot(f(x), x = -10..2, y=-2..2);
```



O método de Newton-Raphson é baseado na seguinte foórmula para o próximo valor na iteração
= - /f ' ( ).

O seguinte procedimento calcula a fórmula para um função .

```
> Newton := proc(f,x) x - f/diff(f,x) end;
```

$$Newton := \mathbf{proc}(f, x)\ \ x - f/diff(f, x)\ \ \mathbf{end\ proc} \tag{11.10}$$

Aplicando o procedimento Newton obtemos a seguinte fórmula

```
> g:= Newton(f(x),x);
```

$$g := x - \frac{e^x - \sin(x)}{e^x - \cos(x)} \tag{11.11}$$

```
> g := unapply(g,x);
```

$$g := x \rightarrow x - \frac{e^x - \sin(x)}{e^x - \cos(x)} \tag{11.12}$$

Escolha um valor inicial para e então a raiz aproximada de usindo:

```
> evalf(g(-5));
```

$$-8.438436205 \tag{11.13}$$

```
> evalf(g(%));
```

$$-9.949828494 \tag{11.14}$$

e assim para obter uma raiz em .

```
> for n from 1 to 10 do; evalf(g(%)); od;
```

$$-9.370628042$$
$$-9.424911633$$
$$-9.424858654$$
$$-9.424858654$$
$$-9.424858654$$
$$-9.424858654$$
$$-9.424858654$$
$$-9.424858654$$
$$-9.424858654$$
$$-9.424858654 \tag{11.15}$$

Cada iteração do procedimento Newton-Raphson pode ser representada graficamente usando o seguinte procedimento **NR**.

```
>  NR:=proc(f,initial,number,left,right,bottom,top)    local
  t,n,N,g,p1,p2:    N:=number:    t:=NULL:    g := unapply
  (Newton(f(x),x),x); x||1:= initial:    for n from 1 to N do:
       t:=t,(x||n,0,x||n,f(x||n)):      x||(n+1):=evalf(g
  (x||n)):    od:
      p1:= plot(f(x),x=left..right,y=bottom..top);
      p2:= plots[pointplot]([t],style=LINE);     print(`Zero
  em x= `,evalf(x||(n)));    plots[display](p1,p2);end:
```

Vejamos um exemplo: aqui -4 é o valor inicial, 5 é o numero de iterações, -6 é o extremo esquerdo do intervalo e 0 é o extremo direito, -1 e 1.5 dão a variação da imagem.

```
> NR(f, -4, 5, -6, -0, -1, 1.5);
```

Zero em x= , -3.183063012



As retas tangentes a curva na aproximação de x dá a próxima aproximação que é onde esta tangente intececta o eixo x .
Isto explica porque
1. O zero encontrado pode não estar perto do nosso valor inicial. Para obter a raiz adjacente ao valor inicial precisamos estar na vizinhança da raiz. Por exemplo:
escolhendo $x_1 = -6$ dá uma raiz em em $x = -6.2813$,
mas
$x_1 = -5$ dá uma raiz em em $x = -94249$.

```
> NR(f, -5, 5, -12, -0, -1, 1.5);
```

Zero em x= , -9.424858654



2. O método de Newton-Raphson falha se a aproximação inicial escolhida é um ponto estacionário.

Por exemplo, tente o procedimento iniciando com .

```
> NR(f, -1.3, 5, -12, -0, -1, 1.5);
```

Zero em x= , −251.3275069



Assim, precisamos conhecer o comportamento geral da função antes de aplicar técnicas numéricas.

11. Integrais de linha - Teorema de Green

1. O teorema de Green

Dado um campo de vetores *F(x,y)= (M,N)* de classe C^1e uma curva *C* simples fechada C^1, o teorema de Green estabelece uma igualdade entre a integral de linha do campo *F* sobre *C* e a integral dupla de
(N_x-M_y) sobre a região limitada pela curva *C*. Onde *C* é orientada positivamente.

Teorema (Green): Seja U um aberto do plano e *C* uma curva simples fechada suave por partes contida em *U* e que encerra uma região *R* do plano. Seja *F(x,y)= (M(x,y), N(x,y))* um campo vetorial de classe em *U*. Então vale a seguinte igualdade:

$$\oint_C M dx + N dy = \iint_R \left(N_x - M_y \right) dA$$

Preste atenção na orientação da curva.

Vamos ver alguns exemplos

```
> restart:
> with(linalg):
> with(plots):
> curva:=[cos(t),sin(t),t=0..2*Pi];
```

$$curva := \left[\cos(t), \sin(t), t = 0\,..2\,\pi\right] \qquad \textbf{(12.1.1)}$$

A parte da variação do parâmetro é muito importante, além da orientação.
Podemos plotar a curva:

```
> plot(curva);
```



Vamos considerar o campo vetorial F(x,y)= (-y,x). Aqui M(x,y)= -y e N(x,y)=x .

```
> vf:=[-y,x];
```

(12.1.2)

$$vf := [-y, x] \tag{12.1.2}$$

```
> F:=fieldplot(vf,x=-2..2,y=-2..2):
> G:=plot(curva):
> display({F,G});
```



Como o campo de vetores é sempre na direção da curva, esperamos que a integral de linha seja positiva.

Para avaliar a integral de linha, nós precisamos substituir a parametrização no campo de vetores, e então integrar o produto interno:

```
> vpar:=subs(x=curva[1],y=curva[2],vf);
```

$$vpar := [-\sin(t), \cos(t)] \tag{12.1.3}$$

```
> Int(dotprod(vpar,diff([curva[1],curva[2]],t)),curva[3]);
```

$$\int_0^{2\pi} \left(\sin(t)\sin(\overline{t}) + \cos(t)\cos(\overline{t})\right) dt \tag{12.1.4}$$

```
> value(%);
```

$$2\pi \tag{12.1.5}$$

Um procedimento em Maple para fazer isto (assumindo **x,y** and **t** com seus papeis usuais) é dado por:

```
> Lineint2:=proc(vf,curve) Int(dotprod(subs(x=curve[1],y=
  curve[2],vf),diff([curve[1],  curve[2]],t)),curve[3]);end;
```

Lineint2 := **proc**(*vf*, *curve*)
Int(*linalg*:-*dotprod*(*subs*($x = curve[1]$, $y = curve[2]$, *vf*), *diff*([*curve*[1], *curve*[2]], *t*)), *curve*[3])
end proc **(12.1.6)**

Testando:

```
> Lineint2(vf,curva);
```

$$\int_0^{2\pi} (\sin(t)\,\sin(\overline{t}) + \cos(t)\,\cos(\overline{t}))\,dt \quad \textbf{(12.1.7)}$$

```
> Int(Int(diff(vf[2],x)-diff(vf[1],y),y=-sqrt(1-x^2)..sqrt(1
  -x^2)),x=-1..1);
```

$$\int_{-1}^{1}\int_{-\sqrt{-x^2+1}}^{\sqrt{-x^2+1}} 2\,dy\,dx \quad \textbf{(12.1.8)}$$

```
> evalf(%);
```

$$6.283185307 \quad \textbf{(12.1.9)}$$

Que é o esperado.

2. Calculando integrais de linha

Neste procedimento, entramos com o campo e a curva e obtemos o valor da integral de linha.
Execute esta worksheet e faça os exemplos.
Sintaxe é linha3d(vf,ch), onde vf é o campo de vetores e ch é o caminho

```
> restart : with(linalg) : with(plots) :
```

```
> linhaint3d :=proc(vf, cv)Int(dotprod(subs(x = cv[1], y = cv[2], z = cv[3], vf),
      diff([cv[1], cv[2], cv[3]], t)), cv[4]) = int(dotprod(subs(x = cv[1], y
      = cv[2], z = cv[3], vf), diff([cv[1], cv[2], cv[3]], t)), cv[4]);end:
```

Exemplos

```
> vf := [2*x, 2*y, 1]; ch := [t*cos(8*t), t*sin(8*t), t, t=0..2];
```

$$vf := [2\,x, 2\,y, 1]$$
$$ch := [t\cos(8\,t), t\sin(8\,t), t, t = 0..2] \quad \textbf{(12.2.1)}$$

```
> linhaint3d(vf,ch);
```

$$\int_0^2 (1 + 2\,t\cos(8\,t)\,\overline{\cos(8\,t) - 8\,t\sin(8\,t)} + 2\,t\sin(8\,t)\,\overline{\sin(8\,t) + 8\,t\cos(8\,t)})\,dt \quad \textbf{(12.2.2)}$$
$$= 6$$

Plotamos o campo de vetores e a curva juntos para obter uma idéia do que esperamos da integral de linha:

```
> F:=fieldplot3d(vf,x=-2..2,y=-2..2,z=-2..2):
> G:=spacecurve(ch):
> display3d({F,G});
```




```
>
```

> vf := [-z * x, y^2 + 2 * x, -x * y]; ch := [-sin(t), cos(t), -t, t =-2 * Pi ..2 * Pi];

$$vf := \left[-z\,x,\ y^2 + 2\,x,\ -x\,y \right]$$

$$ch := \left[-\sin(t),\ \cos(t),\ -t,\ t = -2\,\pi..2\,\pi \right] \tag{12.2.3}$$

```
> linhaint3d(vf,ch);
```

$$\int_{-2\pi}^{2\pi} \left(t\sin(t)\,\cos(\overline{t}\,) - \left(\cos(t)^2 - 2\sin(t)\right)\sin(\overline{t}\,) - \sin(t)\,\cos(t) \right) \mathrm{d}t = 3\,\pi \tag{12.2.4}$$

Plotamos o campo de vetores e a curva juntos para obter uma idéia do que esperamos da integral de linha:

```
> F:=fieldplot3d(vf,x=-2..2,y=-2..2,z=-2..2):
> G:=spacecurve(ch):
> display3d({F,G});
```



>

O pacote de Cálculo Vetorial permite o calculo de integrais de linha. Veja como se faz:

> with(VectorCalculus) :

> SetCoordinates(cartesian$_{x, y}$)

$$cartesian_{x, y} \quad (12.2.5)$$

> LineInt(VectorField(⟨x, y⟩), Line(⟨1, 2⟩, ⟨3, −4⟩))

$$10 \quad (12.2.6)$$

> LineInt(VectorField(⟨x, y⟩), LineSegments(⟨0, 0⟩, ⟨1, 1⟩, ⟨1, −1⟩))

$$1 \quad (12.2.7)$$

> LineInt(VectorField(⟨x^2, y^2⟩), Path(⟨t, t^2⟩, t = 0 ..2))

$$24 \quad (12.2.8)$$

> LineInt(VectorField(⟨y, −x⟩), Circle(⟨0, 0⟩, r))

$$-2\,\pi\,r^2 \quad (12.2.9)$$